C2NOTES

A Calculus and MAPLE 7 Supplement

SM122 **Fall, 2001-2002**

Prepared by:
Professor C. E. Moore
Department of Mathematics
United States Naval Academy

Midshipman:

Section:

Instructor:

Table of Contents

	Review of Differentiation and Anti-differentiation	1
C2M0	Introduction to Maple 7	3
C2M1	Change of Variable	9
C2M2	Rational Fractions or Partial Fraction Decompositions	10
C2M3	Simpson's and Trapezoidal Rule	12
C2M4	Improper Integrals	14
C2M5	Parametric Functions	15
C2M6	Polar Coordinates	16
C2M7	Solutions of Differential Equations	18
	Maple 7 Spreadsheets	19
C2M8	Direction Fields and Euler's Method	22
C2M9	Growth and Decay	25
C2M10	Sequences	26
C2M11	Ratio Test	28
C2M12	Ratio Test for Power Series	29
C2M13	Maclaurin and Taylor Series	30
C2M14	Animation of Taylor Polynomials	32
C2M15	Vectors and the Scalar Product	34
C2M16	Lines, Planes, and Distances in $\mathbb{R}^3$	36
	Summary of Maple Commands	39

Acknowledgments

The author wishes to express his gratitude to the U.S. Naval Academy for the partial support received while writing these notes, and to the following colleagues for their assistance and encouragement, for without their help and expertise these notes would not be accessible on the Internet:

Professor Michael W. Chamberlain

Professor James M. D'Archangelo

Professor Mark D. Meyerson

Also, the author is particularly grateful to Louis at Y & Y Inc. for technical assistance and advice on the use of TEX, the mathematical typesetting system used to produce these notes.

The author wishes to express his considerable appreciation for the assistance and encouragement received from Darren McIntyre of Waterloo Maple Inc.

Calculus II **C2Notes** **Fall, 2000–2001**

by Professor C. E. Moore

Review of Differentiation and Anti-differentiation

As we begin Calculus II it is assumed that the student has covered the derivative of a function in some detail and has learned how to find the anti-derivative of basic functions. This section is provided as a summary of some of these topics. Knowledge of trigonometric functions, exponential functions, and logarithmic functions is assumed, but will be reviewed as necessary throughout the course for reinforcement.

Definition of derivative. Suppose f is defined on an open interval containing x. The derivative of f at x is defined by

$$D_x\left(f(x)\right) = f'(x) \equiv \lim_{h\to 0} \frac{f(x+h)-f(x)}{h}$$

Tangent line. If $f(x_0) = y_0$ and $f'(x_0) = m$, then an equation for the line tangent to the curve $y = f(x)$ is given by

$$y - y_0 = m(x - x_0)$$

Rules of Differentiation Assume that a and b are real numbers and that $f(x)$ and $g(x)$ are differentiable on an open interval containing x.

Rule 1. The derivative is linear. That is, $D_x\big(af(x)+bg(x)\big) = aD_x(f(x)) + bD_x(g(x)) = af'(x) + bg'(x)$.

Rule 2. Product Rule. $D_x\big(f(x)g(x)\big) = D_x(f(x))g(x) + f(x)D_x(g(x)) = f'(x)g(x) + f(x)g'(x)$.

Rule 3. Quotient Rule. On an interval where $g(x) \neq 0$,

$$D_x\left(\frac{f(x)}{g(x)}\right) = \frac{g(x)D_x\big(f(x)\big) - f(x)D_x\big(g(x)\big)}{\big(g(x)\big)^2} = \frac{g(x)f'(x) - f(x)g'(x)}{\big(g(x)\big)^2}$$

Rule 4. Power function derivative. If r is a real number, then

$$D_x(x^r) = rx^{r-1}$$

Examples: $D_x\left(x^{4/3}\right) = \dfrac{4}{3}x^{1/3}$, $D_x\left(\dfrac{1}{x^{4/3}}\right) = -\dfrac{4}{3}\dfrac{1}{x^{7/3}}$, $D_x\left(\sqrt{x}\right) = \dfrac{1}{2}x^{-\frac{1}{2}} = \dfrac{1}{2\sqrt{x}}$

Rule 5. Chain Rule. On an open interval for which $(f\circ g)(x) \equiv f\big(g(x)\big)$ is defined

$$D_x\big((f\circ g)(x)\big) = f'(g(x))g'(x)$$

Example: $D_x\Big(\big(4+x^3\big)^5\Big) = (5)\big(4+x^3\big)^4(3x^2)$

Rule 6. Reciprocal Rule. On an interval where $f(x) \neq 0$ we have

$$D_x\left(\frac{1}{f(x)}\right) = \frac{-f'(x)}{\big(f(x)\big)^2}$$

Derivatives of trigonometric functions. First, we list the derivatives of the three basic functions.

$$D_x(\sin x) = \cos x \qquad D_x(\tan x) = \sec^2 x \qquad D_x(\sec x) = (\sec x)(\tan x)$$

Then we use this information to help determine the derivatives of the cofunctions.

Function	Derivative	Cofunction	Derivative
$\sin x$	$\cos x$	$\cos x$	$-\sin x$
$\tan x$	$\sec^2 x$	$\cot x$	$-\csc^2 x$
$\sec x$	$\sec x \tan x$	$\csc x$	$-\csc x \cot x$

Note that the first column is simply the three basic functions. The second column is the derivative of the first column. The third column is a listing of the cofunctions of the first column. And to form the fourth column, begin by putting a minus sign in front of the expression to follow. Then, simply put the respective cofunctions of the second column after the minus sign.

Example: $D_x\big(\tan^3(\sqrt{x}\,)\big) = 3\tan^2(\sqrt{x}\,)\sec^2(\sqrt{x}\,)\dfrac{1}{2\sqrt{x}}$.

Derivative of exponential and logarithmic functions. $D_x\big(e^x\big) = e^x$ and $D_x(\ln|x|) = \dfrac{1}{x}$, $x \neq 0$. Also, for $a > 0$, $D_x\big(a^x\big) = (\ln a)a^x$.

We will limit our discussion on inverse trigonometric functions to

$$\sin^{-1} x \equiv \arcsin x \qquad \text{and} \qquad \tan^{-1} x \equiv \arctan x$$

We remind the reader that the exponents refer to inverse functions and **not** to reciprocals. The derivatives and corresponding anti-derivatives we will need are listed below:

$$D_x(\arcsin x) = \frac{1}{\sqrt{1-x^2}} \qquad \int \frac{1}{\sqrt{1-x^2}}\,dx = \arcsin x + C$$

$$D_x(\arctan x) = \frac{1}{1+x^2} \qquad \int \frac{1}{1+x^2}\,dx = \arctan x + C$$

Now we turn our attention to anti-derivatives. The easiest place to start is with the power rule.

Anti-derivative of x^r. If $r \neq -1$, then $\displaystyle\int x^r\,dx = \frac{1}{r+1}x^{r+1} + C$.

The anti-derivatives of the basic trigonometric functions are found in the following table:

Function	Anti-derivative	Cofunction	Anti-derivative
$\sin x$	$-\cos x + C$	$\cos x$	$\sin x + C$
$\tan x$	$\ln\lvert\sec x\rvert + C$	$\cot x$	$\ln\lvert\sin x\rvert + C$
$\sec x$	$\ln\lvert\sec x + \tan x\rvert + C$	$\csc x$	$\ln\lvert\csc x - \cot x\rvert + C$

There are other basic trigonometric anti-derivatives that are easily found by reversing the differentiation formulas.

Other trigonometric anti-derivatives.

$$\int \sec^2 x\,dx = \tan x + C \qquad \int (\sec x)(\tan x)\,dx = \sec x + C$$

$$\int \csc^2 x\,dx = -\cot x + C \qquad \int (\csc x)(\cot x)\,dx = -\csc x + C$$

Anti-derivatives of exponential functions. It is assumed that $a > 0$.

$$\int e^x\,dx = e^x + C \qquad \int a^x\,dx = \frac{1}{\ln a}a^x + C$$

Review Exercises: Find derivatives of the following: 1. $x^2\cos x$ 2. $\dfrac{\ln x}{x^2+4}$ 3. $e^{3x}\tan(2x)$

4. Find an equation for the line tangent to the curve $y = \dfrac{x+3}{4+x^2}$ at the point where $x = 2$.

C2M0

Introduction to Maple

Our discussion will focus on Maple 7, which was developed by Waterloo Maple Inc. in Waterloo, Ontario, Canada. Quoting from the *Maple 7 Learning Guide*, “Maple is a *Symbolic Computation System* or *Computer Algebra System*. Both phrases refer to Maple’s ability to manipulate information in a symbolic or algebraic manner. Conventional mathematical programs require numerical values for all variables. By contrast, Maple maintains and manipulates the underlying symbols and expressions, as well as evaluates numerical expressions.” From the second description you see why Maple is designated as a ‘CAS’.

Assignment Format

We are going to begin by establishing a format for each Maple assignment that is to be handed in. Open Maple and obtain a blank worksheet. Do not type the “<” or “>” which are shown to identify your entries. And <Enter> means the “Enter” key. As you begin, the worksheet is in “math mode”, so ‘click’ on the **T** to switch into “text mode”.



For the assignment **C2M1**, type <C2M1> <Enter> and then highlight C2M1 and click on the middle of the three boxes to the right of **B** *I* u so as to center C2M1. The left of these three buttons left-justifies text and the right one right-justifies it. Now,
<down arrow>, then type your name and section as shown.
<Midn Your Name> <Enter>
<Section> <Your section> <Enter>
Having completed this, highlight the three lines and then click on **B** to boldface everything. This is the format you should use for all Maple assignments to be handed in. For example, you should see something like

C2M1

Midn John Doe
Section 1234

Beginning Maple Syntax

Because we are building the foundation for the use of Maple, we will designate this spadework by ♠ when we wish to highlight an important fact. Since getting on-line help is extremely important, we will begin with that. Suppose that you have a question about the command ‘plot’. Then in a worksheet enter <?plot> <Enter> and you will see the information available and links for other related topics.

♠ To obtain on-line help on ‘command’, enter <?command> <Enter>.

You may eliminate the brackets on the left by pressing the function key <F9>. To return to math mode, click on the [>. If we wanted to type a math formula while in text mode we would click on Σ. Later in this section we will discuss *palettes* which allow you to select commands from a menu and avoid using Maple syntax. It is the contention of the author of these notes that learning some Maple syntax is beneficial to the student, so even though you may accomplish the same things by clicking on a symbol, we will show you the syntax that would otherwise be hidden.

In math mode, lines in Maple end with a semicolon or colon.

♠ If a line ends with a semicolon, then the output will be displayed.

♠ If a line ends with a colon, then the display of the output is suppressed.

♠ To activate a line, press <Enter> with the cursor at **any** position on that line. (This does not break the line as it would in a word processor.)

Please type the command lines below in a new worksheet exactly as you see them, remembering to press <Enter>, and note the output. This work is for your benefit and is not intended to be handed in.

```
> a:=4;
> sqrt(a);
> b=4;
> sqrt(b);
```

We did not display the output here because it is important that the reader discover the results for themself. However, the concept is very important. The first line shows the format for assigning a value to the variable 'a'. Think of this as placing the value 4 in a memory location, named 'a', which can be retrieved when needed. The third line is an equation. While it is trivial to solve for b, the value of b is not accessible in this format.

♠ Use `a:=b` to assign the value 'b' to 'a'. Note that 'b' may be a number or an expression.

Expression versus Function

Please enter the following lines in a worksheet, remembering to press <Enter> to activate each line:

```
> A:=x^2+sin(4*x);
> f:=x->sqrt(x^2+9);
> subs(x=4,A);
> f(x); f(4);
> subs(x=4,f(x));
> simplify(%);
```

The first line names the **expression** $x^2 + \sin(4x)$ as 'A'. Note how '$*$' must be used when two factors are to be multiplied. Omitting the $*$ is a common error for beginning Maple users. The second line shows how to define a **function** f. The third line shows how to substitute 4 for x in the expression A. Note: You may **not** use $4 = x$ and expect the same result. This part of the substitution must be of the form 'old' = 'new', because the 'old' will be replaced by the 'new'. Try reversing the order of the substitution by entering <`subs(4=x,A);`>. What difference do you see?

The fourth line above is more important than it appears to be. You see that $f(x)$ is an **expression** and then the fifth line reinforces that. However, were you surprised at the output of the fifth line? There are many levels of operations and simplifications in Maple and substitution does not simplify as far as is possible. The use of % as the `simplify` argument refers to the previous output, which was $\sqrt{25}$.

♠ You must be aware of when the syntax calls for an expression or for a function.

♠ When two factors are to be multiplied, a '$*$' must separate them, i.e. `a*b`.

♠ You may use `%` to refer to the previous output in a Maple command. Warning! See comments about order of execution later in this section.

In that same worksheet enter these command lines and observe the output.

```
> value(Pi^2/6 + sin(Pi/6));
```

$$\frac{1}{6}\pi^2 + \frac{1}{2}$$

```
> evalf(Pi^2/6 + sin(Pi/6));
```

$$2.144934068$$

```
> evalf(Pi^2/6 + sin(Pi/6),75);
```

$$2.14493406684822643647241516664602518921894990120679843773555822937000747041$$

The '`value`' command returns an exact value for an expression, while the `evalf` command converts an exact numerical expression to a floating point number. You also have the option of specifying the number of digits displayed, as shown on the third line. The default is to display ten digits. You may also set the number of digits to be displayed in the remainder of a worksheet by using a command, for example:

```
> Digits:=20;
```

$$Digits := 20$$

```
> evalf(Pi);
```

$$3.1415926535897932385$$

```
> evalf(pi);
```

$$\pi$$

♠ In Maple, `Pi` $= \pi$ is a number, and `pi` $= \pi$ is a small Greek letter with no numerical value.

It will be very useful later to be able to make a function out of an expression. The syntax for this is puzzling. Entering these commands should produce these results:

```
> P:=x^2+cos(x);
```

$$P := x^2 + \cos(x)$$

```
> G:=unapply(P,x);
```

$$G := x \rightarrow x^2 + \cos(x)$$

```
> G(Pi);
```

$$\pi^2 - 1$$

We see that G is a function and that $G(x) = P$.

Maple Graphics

Maple graphics are versatile and easy to use. Let's define $F(x) = 4 - 2x - x^2$ in our worksheet and see how we can get a quick plot of F on $[-2, 3]$. To save space we have included two plots side-by-side. The output of the first is on the left.

```
> F:=x->4-2*x-x^2;
```

$$F := x \rightarrow 4 - 2x - x^2$$

```
> plot(F(x),x=-2..3,color=blue);
```



Plot of F(x)



Plot of P

Note how we used $F(x)$ which is an expression, not just F, in the first plot. Now click on the displayed graph. A box will appear around the graph with small black boxes placed strategically. Move the cursor over the lower righthand box and position it so that a diagonal arrow appears. Click when the arrow is displayed and drag the arrow towards the center of the box, thereby resizing the box. You can change the aspect of the graph by making the box tall or short. You should always make the plots on your homework smaller so as to save paper.

♠ You can resize a Maple plot by clicking on the plot and then dragging the corner of the displayed box.

To plot the function G from above we could use P or $G(x)$ and obtain identical results. The output is on the right above.

```
> plot(P,x=0..Pi,color=magenta);
```

When you wish to plot two functions with the same domain it can be done very easily. However, it is also very easy to confuse this syntax with that of parametric plotting. We will do an example of each so that you will know where to be careful. The placement of the righthand square bracket determines which format you have. In two-dimensional plotting, when you list two expressions and a range inside the square brackets the first function controls the value on the x-axis and the second function controls the value on the y-axis. This is parametric plotting. To save space the output follows on the left.

```
> plot([exp(x),x^2,x=-1..1]);
```





Parametric Plotting **Two Functions**

When you do not include the domain inside the square brackets you get two different plots on the same coordinate system as you can see above on the right. This was produced by:

```
> plot([exp(x),x^2],x=-1..1);
```

♠ To plot the graphs of two functions with the same domain on the same axes, include both functions in square brackets, but exclude the range. If the range is included in the brackets, then the result will be a parametric plot.

♠ The exponential function, e^x, is accessed by `exp(x)` in Maple, which does not recognize e as any particular number. See the following:

```
> evalf(e);
```

$$e$$

```
> evalf(exp(1),90);
```

2.71828182845904523536028747135266249775724709369995957496696762772407663035354759457138217

Maple Packages

Just as specialized mechanics require special tools, and they do not carry every trade's tools with them at all times, Maple has packaged different commands into different libraries so that not all commands need to be put into active memory at all times. Instead, the user may select one or more packages as needed and thereby save computer memory space. For a list of all the packages, see section **3.8** on page 104 of the *Maple 7 Learning Guide*. Or, in a worksheet type <?packages> to see the listing in the *Help* section of Maple.

There are several packages that the beginning calculus student will need. The first is `student`, which contains many calculus operations, and the second is `plots`. As you would expect, `plots` is a graphics package. To invoke, or 'call up' a package you would insert a line in the worksheet before the package is needed such as:

```
> with(plots):
```

If you had ended the line above with a semicolon, then a list of all the commands in `plots` would have been displayed. You should probably try this once just to see what is there. Also, try putting a semicolon after `with(student)` instead of a colon.

More Graphics

How should you plot two functions which have different domains on the same coordinate axes? The answer is to give each plot a name **using a colon at the end to suppress the output** and then display them together. The command `display` is in the graphics package `plots`.

```
> with(plots):
> A:=plot(exp(x),x=-3..3):   ← colon!
```

> `B:=plot(ln(x),x=.01..4):` ← colon!
> `display(A,B);` ← semicolon!



Order of Execution of Commands

Sometimes previous experience with computers causes us to assume that the order of execution of the commands in Maple is the same as the order shown on the screen. Also, we assume that whatever we see in front of us has been executed. Both assumptions are false in Maple. Suppose that you have a worksheet that you had saved previously and that you have just reopened it. You move the cursor down to some line in the worksheet and hit <Enter>. As far as Maple is concerned, this is the first line executed and if it depends on lines above it, then it cannot execute correctly. Computationally, Maple remembers the information in the order in which the lines were executed.

Also, sometimes we want Maple to forget that it has done something and to start over. For this reason the first command line should have `restart:` as the first word. Then, call up the packages that you anticipate using. For example:

> `restart:    with(student):    with(plots):`

is a typical first command line in Calculus I. Since `restart:` clears the memory, what would happen if it occurred last on the line? That's right, the packages would be loaded and then erased from accessible memory.

It is quite reasonable and normal to move around in a worksheet, changing things as needed. In fact, one frequently realizes that a new command line needs to be inserted earlier in a worksheet. To do this, place the cursor on the line above where you want a new line. Then, click on the $\boxed{[>}$ button and a new blank command line will appear. But, when you execute this line by hitting <Enter>, what will be the order of execution of the Maple commands? How do you ensure that the order of execution of the commands is the same as what you see? One approach would be to put the cursor on the first line and hit <Enter> until you reach the last line. That is effective, but it is not necessary. Move the cursor up to 'Edit' on the left at the top of the screen, 'click', and then slide the cursor down to 'Execute'. A side panel will open, slide the cursor over and down to 'Worksheet', and click'. At this point, Maple will execute the worksheet in the order of the commands shown on the screen.

♠ To insert a new command line in a worksheet, put the cursor on the line above and click on $\boxed{[>}$.

♠ **Always** execute a worksheet upon reopening it and before additional commands are added.

♠ **Always** execute a worksheet before saving it and printing it out to be handed in.

Example: Suppose that we are given two points in the plane, $P_1(2,5)$ and $P_2(-1,1)$, and we wish to find the distance between them and an equation for the line that contains them.

> `restart:`
> `x1:=2; y1:=5; x2:=-1; y2:=1;`

$$x1 := 2$$
$$y1 := 5$$
$$x2 := -1$$
$$y2 := 1$$

```
> distance:=sqrt((x2-x1)^2+(y2-y1)^2);
```

$$distance := 5$$

```
> slope:=(y2-y1)/(x2-x1);
```

$$slope := \frac{4}{3}$$

```
> line1:=y-y1=slope*(x-x1);
```

$$line1 := y - 5 = \frac{4}{3}x - \frac{8}{3}$$

```
> y=solve(line1,y);
```

$$y = \frac{7}{3} + \frac{4}{3}x$$

The process is straightforward until we reach the line that begins with `line1`. We assigned the name 'line1' to the equation for the line using the format $y - y_0 = m(x - x_0)$, where m is the slope of the line and $P(x_0, y_0)$ is a known point on the line. We chose to use P_1, but the choice of P_2 would have worked equally well. Then, we established a new equation with y being set equal to the solution of the equation *line* for the variable y. This yields the format $y = mx + b$, which some prefer.

There are two operations that are very basic in calculus, namely differentiation and integration, or anti-differentiation. An expression in x and t, say $\tan(x/t)$, can be differentiated with respect to either variable, so we must remember to specify the variable with respect to which the operation is being performed. Using the expression $P = x^2 + \cos(x)$ from above we have

```
> Pprime:=diff(P,x);
```

$$Pprime := 2x - \sin(x)$$

And if we integrate P

```
> Pint:=int(P,x);
```

$$Pint := \frac{x^3}{3} + \sin(x)$$

There are two commands, `int` and `Int` that are similar, but different. The lower-case produces the value of the integral, while the upper-case produces an unevaluated integral. For example:

```
> A:=int(x^2,x=2..5);
```

$$A := 39$$

```
> B:=Int(x^2,x=2..5);
```

$$B := \int_2^5 x^2\,dx$$

```
> value(B);
```

$$39$$

Now let's do some of the same steps by using a palette. On your command line type <A:=> to get

```
> A:=
```

There are three palettes and to access them you begin by clicking on "View", then "Palettes". If you need symbols, select that palette, but for now we choose "expressions". You should see

Click on the box with the integral symbol $\boxed{\int a}$. Then click on the box with $\boxed{a^b}$. On your command line the cursor appears where you want x inserted, so you type $< x >$, and then **move to the next entry position by using the 'Tab' key**. Enter $< 2 >$, <Tab>, and then the variable of integration, $< x >$ and <Enter>. At the end of the command line put a semicolon and hit <Enter>. This should produce

```
> A:=int(x^2,x);
```

$$\frac{x^3}{3}$$

It is not the purpose of this section to teach you about differentiation or integration. But, you see how that palettes can be used to accomplish these tasks.

Exercises Use Maple to find the plots and answers.

1. Display the graph of $y = 3\sin(2x + \pi/6) + 2$ on the interval $\left[-\pi, \frac{3\pi}{2}\right]$.

2. Define $f(x) = x^2 - 4x + 4$ and $g(x) = \ln(x + 2)$ as functions and in the same plot display their graphs for the interval $[-1, 4]$.

3. Find the value of e^{-2x} for $x = .7$ and display 25 places.

C2M1

Change of Variable

The art of changing the variable in an integration (anti-differentiation) problem must be practiced in order to master it. We learn this valuable tool by trial and error. Maple provides an easy environment in which to try different substitutions. One simple rule to remember when substituting in Maple is that $expr_{old} = expr_{new}$ is the order in which to write the change. Let's begin by doing a simple change of variable for $\int \cos^4(3x)\sin(3x)\,dx$ and then show how it can be done using Maple. Note that reference is made to the **third step** of the substitution. **Students who write out the steps of a substitution carefully, and are meticulous when applying the third step, make far fewer errors than their colleagues who do not. You have been warned!**

Example: $\displaystyle\int \cos^4(3x)\sin(3x)\,dx$ Let's try:

$$u = \cos(3x)$$
$$du = -3\sin(3x)\,dx$$
$$-\frac{1}{3}du = \sin(3x)\,dx \qquad \text{(third step, } -\frac{1}{3}\,du \text{ replaces } \sin(3x)\,dx)$$
$$\int \cos^4(3x)\sin(3x)\,dx = \int (u)^4(-\frac{1}{3}\,du) = -\frac{1}{3}\int u^4\,du$$
$$= -\frac{1}{15}u^5 + C \qquad \text{now resubstitute, using } u = \cos(3x)$$
$$= -\frac{1}{15}\cos^5(3x) + C$$

Maple Example:

After we identify the integral by the name A and display it, we realize that $\cos(3x)$ will become "u". Since the "old" expression precedes the "new" one when doing a substitution in Maple, we write "$\cos(3x) = u$" when applying the Maple command `changevar`. Later when we resubstitute, we will use

"$u = \cos(3x)$". Now consider the Maple worksheet below, paying attention to the order used in the substitutions:

```
> with(student):
> A:=Int((cos(3*x))^4*sin(3*x),x);
```

$$A := \int \cos(3x)^4 \sin(3x)\, dx$$

```
> B:=changevar(cos(3*x)=u,A);
```

$$B := \int -\frac{1}{3} u^4 \, du$$

```
> B:=value(B);
```

$$B := -\frac{1}{15} u^5$$

```
> B:=subs(u=cos(3*x),B);
```

$$B := -\frac{1}{15} \cos(3x)^5$$

C2M1 Problems

These integral problems are to be done two ways. Do them with pencil and paper showing all details and do them with Maple using `changevar`.

1. $\displaystyle\int 24x\left(4+9x^2\right)^5\, dx$ 2. $\displaystyle\int \frac{\cos\sqrt{x}}{\sqrt{x}}\, dx$ 3. $\displaystyle\int \frac{\left(9+\frac{5}{x}\right)^{3/2}}{x^2}\, dx$
4. $\displaystyle\int 12x\sin^{1/2}(3x^2)\cos(3x^2)\, dx$

C2M2

Rational Fractions or Partial Fraction Decompositions

When you were learning basic algebra there were probably problems assigned on adding fractions which had constants, linear expressions, and quadratic expressions in x in the numerators and denominators. Our objective here is to take the answers to those questions and find the fractions that were added together. The need here is to break down a complicated fraction into several simple ones whose anti-derivatives are (much!) easier to find. To approach this systematically we separate the problems into groups determined by the nature of the denominators of the fractions. Recall that $P(x) = x^2 - a^2$ can be factored into $P(x) = (x-a)(x+a)$, so we say that $P(x)$ is *reducible*. Likewise, $Q(x) = x^2 + a^2$ can **NOT** be factored, so $Q(x)$ is *irreducible*. There are different approaches to to these problems and we will use substitution as our method. Two principles from algebra are applicable here. The first states that if two polynomials in x agree for all values of x, then the polynomials have the same coefficients, that is, they are identical. The second states that $x - r$ is a factor of the polynomial $P(x)$ if and only if $P(r) = 0$.

In the following, $P(x)$ is a polynomial in x and the degree of P is less than the degree of the denominator. For cases where the degree of P is greater than or equal to the degree of the denominator, one must divide the polynomials and then work with the remainder. Note that we are forcing the left and right sides to agree for all x, so we may use $\equiv$ instead of $=$.

I. The denominator has distinct linear factors. The expression on the left breaks down as shown:

$$\frac{P(x)}{(x-r_1)(x-r_2)\cdots(x-r_n)} \equiv \frac{a_1}{x-r_1} + \frac{a_2}{x-r_2} + \cdots + \frac{a_n}{x-r_n}$$

and the problem is reduced to determining the coefficients $a_1, a_2, \ldots, a_n$. We are assuming that no two factors are the same.

Example: Decompose $\dfrac{7x-11}{(x-1)(x+1)(x-2)}$

$$
\begin{aligned}
\frac{7x-11}{(x-1)(x+1)(x-2)} &= \frac{a}{x-1}+\frac{b}{x+1}+\frac{c}{x-2} && \text{multiply through by the LHS denominator}\\
7x-11 &= a(x+1)(x-2)+b(x-1)(x-2)+c(x-1)(x+1) && \text{substitute } x=1\\
-4 &= a(2)(-1) \qquad \text{so } a=2 && \text{substitute } x=-1\\
-18 &= b(-2)(-3) \qquad \text{so } b=-3 && \text{substitute } x=2\\
3 &= c(1)(3) \qquad \text{so } c=1 &&
\end{aligned}
$$

From this we conclude:

$$\frac{7x-11}{(x-1)(x+1)(x-2)} = \frac{2}{x-1}-\frac{3}{x+1}+\frac{1}{x-2} \text{ for all values of } x \text{ except } 1,\ -1, \text{ and } 2.$$

II. The denominator has linear factors with repeat(s). In this case, each factor occurs as a denominator up to the power to which it occurs in the original denominator. It is easiest to explain with an example. Suppose the original denominator is $(x-1)^3(x+2)^2(x-4)$ then the decomposition will look like:

$$\frac{a}{x-1}+\frac{b}{(x-1)^2}+\frac{c}{(x-1)^3}+\frac{d}{x+2}+\frac{e}{(x+2)^2}+\frac{f}{x-4}$$

Example: Decompose $\dfrac{3x^2-7x+1}{(x-1)^2(x+2)}$.

$$
\begin{aligned}
\frac{3x^2-7x+1}{(x-1)^2(x+2)} &= \frac{a}{x-1}+\frac{b}{(x-1)^2}+\frac{c}{x+2} && \text{multiply by LHS denominator}\\
3x^2-7x+1 &= a(x-1)(x+2)+b(x+2)+c(x-1)^2 && \text{substitute } x=1\\
-3 &= b(3) && \text{so } b=-1, \text{ substitute that value and simplify}\\
3x^2-7x+1 &= a(x-1)(x+2)-(x+2)+c(x-1)^2 && \text{move } (x+2) \text{ to the other side}\\
3x^2-6x+3 &= a(x-1)(x+2)+c(x-1)^2 && \text{both sides must be divisible by } x-1, \text{ so divide}\\
3x-3 &= a(x+2)+c(x-1) && \text{use } x=1 \text{ again}\\
0 &= 3a \qquad \text{so } a=0 && \text{use } x=-2\\
-9 &= c(-3) \qquad \text{so } c=3 &&
\end{aligned}
$$

And we conclude:

$$\frac{3x^2-7x+1}{(x-1)^2(x+2)} = \frac{-1}{(x-1)^2}-\frac{3}{x+2}$$

III. The denominator has an irreducible quadratic factor. A quadratic factor requires a linear numerator such as $ax+b$.

Example: Decompose $\dfrac{x^2+5}{(x-1)(x^2+2)}$.

$$
\begin{aligned}
\frac{x^2+5}{(x-1)(x^2+2)} &= \frac{a}{x-1}+\frac{bx+c}{x^2+2} && \text{multiply by LHS denominator}\\
x^2+5 &= a(x^2+2)+(bx+c)(x-1) && \text{substitute } x=1\\
6 &= a(3) && \text{so } a=2, \text{ substitute that value and simplify}\\
x^2+5 &= 2x^2+4+(bx+c)(x-1) &&\\
-x^2+1 &= (bx+c)(x-1) && \text{both sides must be divisible by } x-1, \text{ so divide}\\
-x-1 &= bx+c && b=-1,\ c=-1
\end{aligned}
$$

Our decomposition is:

$$\frac{x^2+5}{(x-1)(x^2+2)} = \frac{2}{x-1}-\frac{x+1}{x^2+2}$$

IV. The denominator has a repeated linear factor and an irreducible quadratic factor. The method illustrated here is the easiest one for this case.

Example: Decompose $\dfrac{8x}{(x-1)^2\,(x^2+3)}$.

$$
\begin{aligned}
\frac{8x}{(x-1)^2\left(x^2+3\right)} &= \frac{a}{x-1}+\frac{b}{(x-1)^2}+\frac{cx+d}{x^2+3} && \text{multiply by LHS denominator}\\
8x &= a(x-1)\left(x^2+3\right)+b\left(x^2+3\right)+(cx+d)(x-1)^2 && \text{substitute } x=1\\
8 &= 4b && \text{so } b=2, \text{ substitute that value}\\
8x &= a(x-1)\left(x^2+3\right)+2x^2+6+(cx+d)(x-1)^2 && \text{move } 2x^2+6 \text{ to the LHS}\\
-2x^2+8x-6 &= a(x-1)\left(x^2+3\right)+(cx+d)(x-1)^2 && x-1 \text{ is a factor of RHS and LHS, } \div\\
-2(x-3) &= a\left(x^2+3\right)+(cx+d)(x-1) && \text{substitute } x=1\\
4 &= 4a && \text{so } a=1, \text{ substitute that value}\\
-2x+6 &= x^2+3+(cx+d)(x-1) && \text{move } x^2+3 \text{ to the LHS}\\
-x^2-2x+3 &= (cx+d)(x-1) && \text{divide by } x-1\\
-x-3 &= cx+d && \text{which forces } c=-1 \text{ and } d=-3\\
\frac{8x}{(x-1)^2\left(x^2+3\right)} &= \frac{1}{x-1}+\frac{2}{(x-1)^2}-\frac{x+3}{x^2+3}
\end{aligned}
$$

Now observe how Maple would have done the problem.

Maple Example:

```
> with(student):
> convert(8*x/((x-1)^2*(x^2+3)),parfrac,x);
```

$$\frac{2}{(x-1)^2}+\frac{1}{x-1}-\frac{3+x}{x^2+3}$$

C2M2 Problems

Find the partial fraction decompositions of the expressions using pencil and paper. Then, check your answers using Maple as was done in the preceding Maple Example. Do **NOT** use Maple to duplicate your pencil and paper work.

1. $\dfrac{8x+5}{(x+1)^2\left(x^2+2\right)}$ 2. $\dfrac{4x^2+4x+12}{x^2\left(x^2+4\right)}$ 3. $\dfrac{7x^2-17x+1}{(x-2)^2\left(x^2+1\right)}$ 4. $\dfrac{x^3-2x^2+2x-1}{x^3\left(x^2+1\right)}$

C2M3

Simpson's and Trapezoidal Rule

Riemann sums, Simpson's Rule, and the Trapezoidal Rule are available in Maple in the Student package. The example chosen here involves the sine function on the interval $[1,3]$ using 60 subintervals for the sums and 10 subintervals for the graphics. Because we wanted the decimal or floating point answer we used `evalf` instead of `value`, which would have listed a long summation. The actual integral is $\int_1^3 \sin x\,dx = -\cos 3+\cos 1 \approx 1.530294803$. Please observe the output of each command line below.

```
> restart:    with(student):
> simpson(sin(x),x=1..3,60);
```

$$\frac{1}{90}\sin(1)+\frac{1}{90}\sin(3)+\frac{2}{45}\left(\sum_{i=1}^{30}\sin\left(\frac{29}{30}+\frac{1}{15}i\right)\right)+\frac{1}{45}\left(\sum_{i=1}^{29}\sin\left(1+\frac{1}{15}i\right)\right)$$

```
> evalf(%);
```

$$1.530294813$$

```
> trapezoid(sin(x),x=1..3,60);
```

$$\frac{1}{60}\sin(1)+\frac{1}{30}\left(\sum_{i=1}^{59}\sin\left(1+\frac{1}{30}i\right)\right)+\frac{1}{60}\sin(3)$$

```
> evalf(%);
```

$$1.530153105$$

```
> Int(sin(x),x=1..3);
```

$$\int_1^3 \sin(x)\,dx$$

```
> value(%);
```

$$-\cos(3)+\cos(1)$$

```
> evalf(%);
```

$$1.530294803$$

Accuracy The error estimates for the Trapezoidal Rule and Simpson's Rule are stated in the course textbook. As a reminder, if $|f''(x)| \le K$ and $|f^{(iv)}(x)| \le M$, and n subintervals are used, then the errors for the respective rules, E_T and E_S, satisfy

$$|E_T| \le \frac{K(b-a)^3}{12\,n^2} \qquad \text{and} \qquad |E_S| \le \frac{M(b-a)^5}{180\,n^4}$$

when applied over the interval $[a,b]$.

Trapezoidal Rule Maple Example Approximate $\displaystyle\int_0^{\pi/3} \sin(2x)\,dx$ to within $\frac{1}{1000}$ using the Trapezoidal Rule. Determine the number of subintervals necessary to achieve the requested accuracy by applying the estimate displayed above. It is best if our second derivative is a function, so we use `unapply` to create a function from an expression.

```
> restart:    with(student):
> f:=x->sin(2*x);
```

$$f := x \to \sin(2x)$$

```
> f2:=unapply(diff(f(x),x,x),x);
```

$$f2 := x \to -4\sin(2x)$$

At this point we have the second derivative of f as a function. We must find the maximum value of the absolute value of the second derivative on the interval. There are several ways to do this, graphically, knowing something about the function which makes a bound obvious, and by comparing values using small increments. While in general the graphical method is simplest, here we know that $\sin(\alpha)$ and $\cos(\alpha)$ are bounded by ±1, so that $|f''(x)| \le 4$. The second option is to plot $|f''(x)|$ and estimate an upper bound, K. Then, plot both $|f''(x)|$ and K on the same axes to see if your estimate is reasonable.

```
> plot(abs(f2(x)), x=0..Pi/3);
```



Based on the graph, choose $K=4$.

```
> K:=4;
```

$$K := 4$$

Equate the error and the overestimate and solve for the value of n that works.

```
> Eqn1:=(Pi/3-0)^3*K/(12*n^2)=1/1000;
```

$$Eqn1 := \frac{1}{81}\frac{\pi^3}{n^2} = \frac{1}{1000}$$

```
> solve(Eqn1,n);
```

$$\frac{10}{9}\sqrt{10}\,\pi^{(3/2)}, -\frac{10}{9}\sqrt{10}\,\pi^{(3/2)}$$

```
> evalf(%);
```

$$19.56511025, -19.56511025$$

Since n must be an integer, choose $n = 20$

```
> app:=trapezoid(f(x),x=0..Pi/3,20);
```

$$app := \frac{1}{120}\pi\left(2\left(\sum_{i=1}^{19}\sin\left(\frac{1}{30}i\pi\right)\right)+\frac{1}{2}\sqrt{3}\right)$$

```
> approx:=evalf(app);
```

$$approx := .7493144853$$

```
> ans:=evalf(int(f(x),x=0..Pi/3));
```

$$ans := .7500000000$$

```
> abs(approx-ans);
```

$$.0006855147$$

So, we have achieved the requested accuracy.

Before we leave this problem, let's consider another way to locate K. Basically, we are going to divide up the interval into a large number of small subintervals and check the value of $|f''(x)|$ on each and compare values. Let's suppose that we want to use 100 subintervals.

```
> n1:=100;  a:=0; b:=Pi/3;
```

$$n1 := 100$$
$$a := 0$$
$$b := \frac{\pi}{3}$$

```
> del:=(b-a)/n1;
```

$$del := \frac{1}{300}\pi$$

```
> S:=seq(abs(f2(a+i*del)),i=0..n1):
> K:=max(S);
```

$$K := 4$$

C2M3 Problems:

1. Use Maple to find the requested approximations. $\int_0^2 \sqrt{1+x^2}\,dx$, $n = 40$, use `simpson, trapezoid`

2. Modify the Maple Example above and use Simpson's Rule instead of the Trapezoidal Rule to approximate $\int_0^2 \frac{x^2}{1+x^4}\,dx$ to within $\frac{1}{1000}$. Use the sequence method to determine a value for M.

3. Modify the Maple Example above and then use Simpson's Rule instead of the Trapezoidal Rule to approximate $\int_0^1 x\arctan(x)\,dx$ to within $\frac{1}{10000}$. Determine the value of M graphically.

C2M4

Improper Integrals

Improper integrals can occur in two different ways. The interval of integration can be unbounded, or the integrand can be an unbounded function. As you know, in the respective cases

$$\int_a^\infty f(x)\,dx \equiv \lim_{M\to\infty}\int_a^M f(x)\,dx$$

$$\int_a^b f(x)\,dx \equiv \lim_{t\to a^+}\int_t^b f(x)\,dx$$

where, in the second case f is unbounded at a. In Maple, we will evaluate the integral from a to M and then evaluate the limit of that result as the definition suggests in order to reinforce the concepts.

Maple Example:

Evaluate the improper integral $\int_1^\infty e^{-2x}\,dx$

We find an expression $A(M)$ which expresses the area of the region under the curve as a function of the righthand endpoint, M. Then we evaluate the limit

$\lim_{M\to\infty} A(M)$




```
> with(student):
> A:=Int(exp(-2*x),x=1..M);
```

$$A := \int_1^M e^{(-2x)}\,dx$$

```
> A:=value(A);
```

$$A := -\frac{1}{2}e^{(-2M)} + \frac{1}{2}e^{(-2)}$$

```
> limit(A,M=infinity);
```

$$\frac{1}{2}e^{(-2)}$$

Maple Example: Evaluate the improper integral $\int_0^1 \frac{1}{x^{1/3}}\,dx$ Note that the integrand is discontinuous at the lefthand endpoint.

```
> B:=Int(x^(-1/3),x=t..1);
```

$$B := \int_t^1 \frac{1}{x^{1/3}}\,dx$$

```
> B:=value(B);
```

$$B := \frac{3}{2} - \frac{3}{2}t^{2/3}$$

```
> limit(B,t=0);
```

$$\frac{3}{2}$$

C2M4 Problems Evaluate the improper integrals using Maple.

1. $\int_2^\infty \frac{1}{x^{4/3}}\,dx$ 2. $\int_3^\infty \frac{\ln x}{x}\,dx$ 3. $\int_0^8 x^{-2/3}\,dx$ 4. $\int_0^1 x\ln x\,dx$
5. $\int_0^1 \frac{e^{\sqrt{x}}}{\sqrt{x}}\,dx$

C2M5

Parametric Functions

Have you ever played with a toy called *"Etch-a-Sketch"*? One hand controls the x-axis while the other controls the y-axis. It is as if you are graphing $(x(t), y(t))$, $a \le t \le b$, which is exactly what happens when a function in the plane is defined parametrically. Be very careful where you place the right bracket,], when using Maple to plot parametric graphs.

Maple Example: Plot $x(t) = \sin(13t)$, $y(t) = \cos(7t)$ for $0 \le t \le 6\pi$ which produces a *lissajou.* The plot is on the left below. As you can see, the scaling is a little off because the "square" is two units on each side. For a little fun, increase the coefficients to say 43 and 37 and see what happens. You may also wish to increase the domain.

```
> plot([sin(13*t),cos(7*t),t=0..6*Pi],color=navy);
```





Maple Example: Ellipses are easy this way. Plot $\dfrac{x^2}{3^2} + \dfrac{y^2}{2^2} = 1$. The Maple output is above on the right. When you have $\dfrac{x^2}{a^2} + \dfrac{y^2}{b^2} = 1$ you may plot this by using $x(t) = a\,\cos(t)$ and $y(t) = b\,\sin(t)$ for $0 \le t \le 2\pi$. So,

```
> plot([3*cos(t),2*sin(t),t=0..2*Pi]);
```

C2M5 Problems Use Maple to display the parametric graphs of the given functions.

1. $x = e^t,\ y = e^{2t},\ -1 \le t \le 2$
2. $x = 2\sec t,\ y = \tan t,\ -\pi/2 < t < \pi/2$
3. $x = t - \sin t,\ y = 1 - \cos t,\ 0 \le t \le 4\pi$
4. $x = \cos^3 t,\ y = \sin^3 t,\ 0 \le t \le 2\pi$

C2M6

Polar Coordinates

For many graphs in polar coordinates the easiest place to start is with a rectangular plot. We will call these *helper diagrams.* For the function $r = 2\sin(3\theta)$ we have the rectangular plot on the left and the polar plot on the right. We can see from the helper diagram that as θ varies from 0 to $\pi/3$ the radius will start at 0, reach 1 at $\pi/6$, and decrease back to 0 at $\pi/3$. The next thing we observe is that from $\pi/3$ to $2\pi/3$ the radius will take on negative values. At each point where the helper diagram attains the value 2, the polar graph will be two units from the origin for that angle. And, whenever the helper diagram has the value of -2, the polar graph will be two units in the opposite direction of that angle. At points on the helper diagram where the value is 0, the polar graph will be tangent to that angle as the graph passes through the polar origin. To sketch the polar plot, begin by drawing a circle of radius 2, then draw radial lines for multiples of $\pi/6$. Mark the intersection for each peak and valley of the helper diagram. These have polar coordinates whose angles correspond to odd multiples of $\pi/6$. Remember, sometimes it is a 2 and sometimes it is a radius of -2. On the polar graph, put your finger at the origin and trace the plot, realizing that you are going to go around the plot twice.





Plotting in polar coordinates is very easy using Maple. We need the library `plots` in order to use `polarplot`. We will use t instead of θ as the variable sometimes.

Maple Example 1: Plot $r = 2\sin(3\theta)$. On the right, $r = 2\cos(3\theta)$ is displayed so that you may compare.

```
> with(plots):
> polarplot(2*sin(3*t),t=0..2*Pi,color=green);
```





Maple Example 2: Polarplot may also be used parametrically. We will define two plots and give them names, *A1* and *B1*, and end their lines with colons to suppress the output. Using `polarplot` parametrically, the first coordinate determines r and the second determines θ. Each may be functions of a third variable whose domain must be specified within the square brackets.

Maple Example 3:

```
> with(plots):
> A1:=polarplot([r,Pi/3,r=1..Pi],color=blue):
> B1:=polarplot([2,theta,theta=-Pi/3..3*Pi/4]):
> display({A1,B1});
```



Maple Example 4: Suppose we consider the vertical line $x = 2$. We know that $x = r\cos(\theta)$ in polar coordinates, so set the x values equal. This means that $r\cos(\theta) = 2$ and that we may solve for r, which produces

$$r = \frac{2}{\cos(\theta)} = 2\sec(\theta)$$

We must avoid dividing by 0, so odd multiples of $\pi/2$ must be avoided. Here we will stay between $-\pi/3$ and $\pi/3$.

```
> with(plots):
> polarplot([2*sec(t),t,t=-Pi/3..Pi/3]);
```



C2M6 Problems Use Maple to determine the polar graphs of the given functions. Remember, you must set the domain. Also, resize the output to a reasonable size. Save paper.

1. $r = 1 + \cos(\theta)$ 2. $r = 2\csc(\theta)$ 3. $r = \sqrt{2} - 2\cos(\theta)$ 4. $r = \theta,\ -\pi \le \theta \le 2\pi$

C2M7

Solutions of Differential Equations

A differential equation arises when there is a relationship involving a function and one or more of its derivatives. For example

$$y'' + 5y' + 6y = 0$$

is such an equation. A function is a solution of this equation if you obtain 0 when you add its second derivative to 5 times its first derivative and then add 6 times the function itself.

Maple Example 1 Use Maple to verify that $y(t) = ae^{-3t} + be^{-2t}$ is a solution of the differential equation shown above, where a and b are arbitrary constants.

```
> with(student):
> de1:={diff(y(t),t,t)+5*diff(y(t),t)+6*y(t)=0};
```

$$de1 := \{\left(\frac{\partial^2}{\partial t^2} y(t)\right) + 5\left(\frac{\partial}{\partial t} y(t)\right) + 6\, y(t) = 0\}$$

```
> y1:=a*exp(-3*t)+b*exp(-2*t);
```

$$y1 := ae^{(-3t)} + be^{(-2t)}$$

```
> eval(de1,y(t)=y1);
```

$$\{0 = 0\}$$

which shows that for any constants a and b, $y(t)$ is a solution of the given equation.

Maple Example 2 Determine whether $y(x) = e^x + ce^{-2x}$ is a solution of

$$y' + 2y = 3e^x$$

for any value of the constant c.

```
> de2:={diff(y(x),x)+2*y(x)=3*exp(x)};
```

$$de2 : \{\left(\frac{\partial}{\partial x} y(x)\right) + 2y(x) = 3e^x\}$$

```
> y2:=exp(x)+c*exp(-2*x);
```

$$y2 := e^x + ce^{(-2x)}$$

```
> eval(de2,y(x)=y2);
```

$$\{3e^x = 3e^x\}$$

How would we know if we did not have a solution? let's define a different function and see what happens.

```
> y3:=2*exp(x)+C*exp(-2*x);
```

$$y3 := 2e^x + Ce^{(-2x)}$$

```
> eval(de2,y(x)=y3);
```

$$\{6e^x = 3e^x\}$$

Now in order for $y3$ to be a solution, the last equation, $6e^x = 3e^x$, would have to be true for every x. But this is true for *no* x, so $y3$ is not a solution.

C2M7 Problems: Use Maple and the method illustrated above to determine whether the given function is a solution of the differential equation.

1. $y = \sin x + x^2$, $\quad y'' + y = x^2 + 2$
2. $y = e^{2x} - 3e^{-x}$, $\quad y'' - y' - 2y = 0$
3. $x = 2e^{3t} - e^{2t}$, $\quad \dfrac{d^2x}{dt^2} - x\dfrac{dx}{dt} + 3x = -2e^{2t}$
4. $x = \cos 2t$, $\quad \dfrac{dx}{dt} + tx = \sin 2t$
5. $x = \cos t - 2\sin t$, $\quad x'' + x = 0$

Maple 7 Spreadsheets

In Maple 7 the user may insert a spreadsheet into the Maple worksheet. One characteristic of these spreadsheets that must be noted early is that changing an entry does not cause the spreadsheet to immediately recalculate as is the case in the standard spreadsheet. This is easily done by clicking on a button, however. The reader is reminded that the easiest way to get help or information about spreadsheets in Maple is to enter a worksheet and type in

> `?spreadsheet` <enter>

and a list of topics is displayed. The reader is warned that there is an error in the identification of row and column headers in the section that tries to explain which is which. Column headers are listed in a row as A, B, C, etc. while row headers are listed next to the first column as 1, 2, 3, etc. This is reversed in the diagram you would see in the help display.

Please open a blank worksheet and enter the following lines:

```
> with(student):
> f:=x->2+2*x-x^2;
```

$$f := x \to 2 + 2x - x^2$$

```
> g:=x->sin(2*x);
```

$$g := x \to \sin(2x)$$

With the mouse, move the cursor arrow to the command **Insert**, click and then click on **Spreadsheet**. In addition to the material already entered, you should see a blank spreadsheet ready to be resized to fit your needs. Before we do that, note that the menu bars above have changed and that they look like this:



Before you inserted the spreadsheet, you could not select **Spreadsheet**, but now it is an option. Click on it and check out the menu. Note the four boxes on the left and below the main bar. The one to the left is very useful. When you need to fill cells in a spreadsheet in some direction, click on this box. More specific instructions follow when this process is needed in our example.

Now we are ready to resize the blank spreadsheet. Outside the spreadsheet, but near the lower righthand corner, click the mouse. A drag box outline should appear around the spreadsheet. Then, move the cursor to that corner until the diagonal arrow appears. Move that corner down and to the left so that you end up with five columns and 15 or so rows. This might take several tries. When you have done this, click on the cell in row 1 and column A. In the first row, enter *x*, *f(x)*, *x*, and *g(x)*. The arrow keys will move from one cell to another. You will note that the value of the function has appeared where you entered $f(x)$ and $g(x)$. We are going to illustrate different methods to accomplish the same things. In cell **A2** enter 0 and move to cell **B2**. One option here is to enter `f(~A2)` and we will do that. The other is to enter `eval(~$B$1,x=~A2)`. The dollar signs mean that cell **B1** will always be used and not just refer to the cell directly above. This is an *absolute* reference rather than a *relative* one.

Our objective now is to fill in the two columns, **A** and **B**. Click on **A2** and highlight down to cell **A12**. The menus at the top of the screen change when you are in a spreadsheet. At the extreme left and on the third row of the menus, you will find a button that looks like three window panes and the shade is pulled down in the top one. It also has an arrow pointing down. Click on this button and a menu pops up. Click on the window that indicates 'step size' and enter .1 , then click on **OK**. You should see the first column of the completed spreadsheet. Move to cell **B2** and highlight down to **B12**. On the top line of the menus click **Spreadsheet**, click on **Fill - Down**. You don't actually click on **Fill** because when the arrow touches it, the side menu with **Down** pops up immediately. This should complete the second column.

Before continuing with the spreadsheet entries, move the cursor outside of the spreadsheet to the command line above the spreadsheet and hit <Enter>. This should cause a command line to appear below the spreadsheet. Enter $M :=$ on that line and then highlight the cells in the first two columns from **A2** to **B12**.

Click on the **Copy** button, move the cursor to after the $M :=$ on the line below, and click on the **Paste** button. Immediately put a colon at the end of the line and then <Enter>. The colon suppresses the output, which is a matrix of 2×2 matrices. On the next line enter `c:=convert(M,‘list‘)[1];` , being careful to use a left single quote on both sides of *list*, and a list of those 2×2 matrices is shown. We can use this list in an interesting way. Enter `plot(c,style=line);` and <Enter>, and a plot appears. It is rather smooth because our x values are close together.

Let's return to the spreadsheet and put 0 in cell **C2**. Move down to cell **C3** and enter `~C2+Pi/12`. Then, highlight from **C3** to **C12** and click on **Spreadsheet, Fill - Down**. In cell **D2** enter `eval(~$D$1,x=~C2)` and 0 appears. Highlight **D3** down to **D12**, then click on **Spreadsheet, Fill - Down**. This completes the spreadsheet.

Here is the tricky part. To plot the second function, highlight the rectangular area **C2** to **D12**, copy and paste to the command line below the spreadsheet, insert
`M1:=` before `MATRIX`,
a colon `:` at the end of the last line, and <Enter>.
On the next line put
`c1:=convert(M1,‘list‘)[1];` <Enter>
and follow that line with
`plot(c1,style=line);` and <Enter>,
and a polygonal graph appears. The spreadsheet is on the right and the remainder of the worksheet follows.

	A	B	C	D
1	x	$2+2x-x^2$	x	$\sin(2x)$
2	0	2	0	0
3	.1000	2.1900	$\frac{1}{12}\pi$	$\frac{1}{2}$
4	.2000	2.3600	$\frac{1}{6}\pi$	$\frac{1}{2}\sqrt{3}$
5	.3000	2.5100	$\frac{1}{4}\pi$	1
6	.4000	2.6400	$\frac{1}{3}\pi$	$\frac{1}{2}\sqrt{3}$
7	.5000	2.7500	$\frac{5}{12}\pi$	$\frac{1}{2}$
8	.6000	2.8400	$\frac{1}{2}\pi$	0
9	.7000	2.9100	$\frac{7}{12}\pi$	$-\frac{1}{2}$
10	.8000	2.9600	$\frac{2}{3}\pi$	$-\frac{1}{2}\sqrt{3}$
11	.9000	2.9900	$\frac{3}{4}\pi$	-1
12	[illegible]	3	$\frac{5}{6}\pi$	$-\frac{1}{2}\sqrt{3}$

```
> M:=MATRIX([[0, 2], [.1, 2.19], [.2, 2.36], [.3, 2.51], [.4, 2.64], [.5, 2.75],
  [.6, 2.84], [.7, 2.91],[.8, 2.96], [.9, 2.99], [1, 3]]):
```

```
> c:=convert(M,‘list‘)[1];
```

c := [[0, 2], [.1, 2.19], [.2, 2.36], [.3, 2.51], [.4, 2.64], [.5, 2.75], [.6, 2.84], [.7, 2.91], [.8, 2.96], [.9, 2.99], [1, 3]]

```
> plot(c,style=line);
```



```
> M1:=MATRIX([[1/12*Pi, 1/2], [1/6*Pi, 1/2*sqrt(3)], [1/4*Pi, 1], [1/3*Pi, 1/2*sqrt(3)],
  [5/12*Pi, 1/2],[1/2*Pi, 0], [7/12*Pi, -1/2], [2/3*Pi, -1/2*sqrt(3)], [3/4*Pi, -1],
  [5/6*Pi, -1/2*sqrt(3)]]):
```

```
> c1:=convert(M1,'list')[1];
```

$$c1 := [[\frac{1}{12}\pi, \frac{1}{2}], [\frac{1}{6}\pi, \frac{1}{2}\sqrt{3}], [\frac{1}{4}\pi, 1], [\frac{1}{3}\pi, \frac{1}{2}\sqrt{3}], [\frac{5}{12}\pi, \frac{1}{2}], [\frac{1}{2}\pi, 0], [\frac{7}{12}\pi, \frac{-1}{2}], [\frac{2}{3}\pi, \frac{-1}{2}\sqrt{3}], [\frac{3}{4}\pi, -1], [\frac{5}{6}\pi, \frac{-1}{2}\sqrt{3}]]$$

```
> plot(c1,style=line);
```



Do you remember the array of trigonometric derivatives in the Review section at the beginning of these notes? It should be easy to duplicate using a Maple spreadsheet. Open a spreadsheet and in the first row insert *function*, *derivative*, *cofunction*, *derivative*. In the first column, **A**, starting with **A2**, enter *sin(x)*, *tan(x)*, and *sec(x)*. Move to **B2** and enter `diff(~A2,x)`. Highlight **B2** to **B4**, then click on **Spreadsheet, Fill - Down** successively. Enter the cofunctions in Column **C**, and the derivatives in column **D** similar to how you handled column **B**.

	A	B	C	D
1	*function*	*derivative*	*cofunction*	*derivative*
2	$\sin(x)$	$\cos(x)$	$\cos(x)$	$-\sin(x)$
3	$\tan(x)$	$1+\tan(x)^2$	$\cot(x)$	$-1-\cot(x)^2$
4	$\sec(x)$	$\sec(x)\tan(x)$	$\csc(x)$	$-\csc(x)\cot(x)$

Let's consider one more example. Start a worksheet, define a function $f(x) = \cos(2x)$, and set $a = Pi/6$ Open a spreadsheet and in cell **A1** put $f(x)$. Move to **A2** and enter `diff(~A1,x)`. Highlight **A2** down to **A8**. Click on **Spreadsheet, Fill - Down** and note that succesive derivatives appear. In **B1**, put `eval(~A1,x=a)` and then highlight **B1** down to **B8**. Click on **Spreadsheet, Fill - Down**. When you reach the section on Taylor series you will realize just how useful a spreadsheet like this example can be.

C2M8

Direction Fields and Euler's Method

Direction Fields As we begin our study of differential equations we come across equations of the form $y' = f(x,y)$, which means that at each point in some region the slopes of solutions of the equation are known. It is very useful to be able to plot short line segments at a reasonable number of points and thereby have a picture of the behavior of these slopes. The command we need in Maple is found in the package 'plots' and the problem is that the input must have two coordinates, not one. We can get around this by making the first coordinate equal to 1 and the second the value of $f(x,y)$.

Maple Example: Suppose we are given a direction field $y' = x + y$ and we wish to plot it.

```
> with(plots):
> f:=(x,y)->x+y;
```

$$f := (x,y) \to x + y$$

```
> F:=(x,y)->[1,f(x,y)];
```

$$F := (x,y) \to [1, f(x,y)]$$

```
> fieldplot(F(x,y),x=-1..2,y=0..4);
```



```
> with(plots):
> f:=(x,y)->x+y;
```

$$f := (x, y) \to x + y$$

Click on **Insert** and **Spreadsheet** and resize the spreadsheet so that there are 3 columns and 12 rows. In the first row insert *n, xn, yn*, move to the second row and insert 0, 0, 1. That last value is $y(0)$.
First column. Highlight the first column from **A2** to **A12**, and then click on the button on the extreme left of the third row of the menus at the top of the page. The button looks like a window with three vertical panes and the blind pulled down over the top pane, along with an arrow pointing down. In the menu box that appears, insert a step size of 1 and a stop value of 10.

Second column. Highlight the second column from **B2** to **B12**, click on the button above on the extreme left (same as before), and insert a step size of .1 and a stop value of 1.

Third column. Carefully insert into **C3**: `~C2+(.1)*f(~B2,~C2)` and <Enter>. Note how this entry will reflect the old value of y plus the slope times the change in x, as it should. To complete this column, highlight from **C3** to **C12**, click on **Spreadsheet**, **Fill**, and **Down**. The spreadsheet should look like the one displayed below.

	A	B	C
1	*n*	*xn*	*yn*
2	0	0	1
3	1	.1000	1.1000
4	2	.2000	1.2200
5	3	.3000	1.3620
6	4	.4000	1.5282
7	5	.5000	1.7210
8	6	.6000	1.9431
9	7	.7000	2.1974
10	8	.8000	2.4871
11	9	.9000	2.8158
12	10	1	3.1874

There should be a prompt > below the spreadsheet. Enter `M:=` and then highlight the region in the spreadsheet from **B2** to **C12**, click on the **Copy** button at the top, move the cursor to after the `M:=` on the line below and then click on the **Paste** button at the top. Insert a colon at the end of that line, and then <Enter>. You should see:

```
> M:=MATRIX([[0, 1], [.1, 1.1], [.2, 1.22], [.3, 1.362], [.4,
1.5282], [.5, 1.72102], [.6, 1.943122], [.7, 2.1974342], [.8,
2.48717762], [.9, 2.815895382], [1, 3.187484920]]):
```

Now we must pair the points together by converting this *matrix* of 1×2 matrices to a *list* of 1×2 matrices. Then we will be able to treat each 1×2 matrix as a point in the plane and plot the graph.

```
> a:=convert(M,'list')[1];
```

$$a := [[0, 1], [.1, 1.1], [.2, 1.22], [.3, 1.362], [.4, 1.5282], [.5, 1.72102], [.6, 1.943122], [.7, 2.1974342], [.8, 2.48717762], [.9, 2.815895382], [1, 3.187484920]]$$

We are going to plot these points and join them together with line segments, and then display the plot at the same time that we display the direction field. This means that we must give these plots names and suppress their output until we are ready.

```
> A:=pointplot(a,style=line,color=blue):
```

Now to construct a function that Maple can interpret as acceptable for the command `fieldplot`:

```
> F:=(x,y)->[1,f(x,y)];
```

$$F := (x, y) \to [1, f(x, y)]$$

```
> B:=fieldplot(F(x,y),x=-1..2,y=0..4):
```

Because we happen to know the solution to this equation we will include it in the plot.

```
> C:=plot(2*exp(x)-x-1,x=-1..1.2,color=red):
> display(A,B,C);
```



C2M8 Problem Use Maple to graph an approximation to the solution of the differential equation

$$y' = x - y^2 \qquad y(1) = -1$$

Use a step size of 0.1 and approximate $y(3)$. This requires twenty steps. The work you submit should reflect the Maple Example above and should display the spreadsheet and display the approximation on the same plot with the direction field. Do not be concerned about the precise solution.

C2M9

Growth and Decay

The exponential function appears in many problems that involve physical phenomena. Newton's **Law of Cooling**, growth and decay, and the mixing of solutions come to mind as examples. In each case, the rate of change of some quantity is proportional to some aspect of the amount present. Imagine placing one steel bar at 32° Fahrenheit in a room that is 40° and another in a room that is 100°. Certainly the temperature of the second bar will change more rapidly than that of the first. Newton's **Law of Cooling** asserts that the rate of change of the surface temperature of an object is proportional to the difference between that temperature and that of the surrounding medium. So if F is the temperature of the steel bar and R_T is the room temperature, then

$$\frac{dF}{dt} = k\,(F - R_T)$$

Example: Suppose that a steel bar is cooled to $-10°$ F and it is placed in a room that is 70° F. Ten minutes later the temperature of the bar is 25°. Find an expression for the temperature at any time t and specifically at $t = 30$. By column

$\frac{dF}{dt} = k(F-70)$
$\frac{dF}{F-70} = k\,dt$
$\int \frac{dF}{F-70} = \int k\,dt$
$\ln|F-70| = kt + C$ but $|F-70| = 70-F$
$70 - F = e^{kt+C}$ $F(0) = -10 \Rightarrow C = \ln 80$
$F = 70 - e^{kt+\ln 80} = 70 - 80e^{kt}$

$F(10) = 25 = 70 - 80\,e^{10k}$
$80\,e^{10k} = 70 - 25 = 45$
$e^{10k} = \frac{45}{80} = \frac{9}{16}$
$10k = \ln\frac{9}{16}$
$k = -.0575364$
$F = 70 - 80e^{-.0575364\,t}$

which yields $F(30) = 55.7617°$. The exponential term goes to zero as time increases, so let's use Maple to graph F and see how F tends to 70.

```
> with(plots):
> F:=70-80*exp(-.0575364*t);
```

$$F := 70 - 80\,e^{(-.0575364\,t)}$$

```
> A:=plot(F,t=0..50):
> B:=plot(70,t=0..50):
> display(A,B);
```



Consider how the slope of the tangent line will change as a point moves from left to right along the curve. This means that the rate of change is a decreasing function, which is consistent with Newton's **Law of Cooling**.

C2M9 Problems: In problems 1 and 2, use Maple to plot the graph(s) of the given function(s) on the same axes. Remember that the exponential function e^x in Maple is `exp(x)`.

1. $f(t) = 60 + 30\,e^{-.3\,t}$ $g(t) = 60,$ $0 \le t \le 40$
2. $h(t) = 3\,e^{-t},\ j(t) = -4\,e^{-3\,t},\ h(t) + j(t),\ 0 \le t \le 6$
3. (Pencil and paper) A steel bar is heated to 200° and placed in a room whose temperature is 60°. Thirty minutes later the bar is 110°.

   (a) Find an expression for the temperature of the bar at any time t.

   (b) At what time will the temperature of the bar be 80°?

C2M10

Sequences

The concept of *limit* is as basic as it gets in calculus. And, to understand how a limit works one must learn to recognize two ingredients, *accuracy* and *control.* Intuitively, when our input gets close to one value our output will get close to another, which we call the limit. So, we must have a means of discussing "getting close to". A *sequence* is a function whose domain is the natural numbers, or a subset thereof, and whose range is the real numbers, $\mathbb{R}$. We can also have sequences of integrals, matrices, or complex numbers, but real numbers will suffice for now.

Notation: $\{a_n\} = \{a_1, a_2, a_3, \ldots, a_n, \ldots\}$ denotes a sequence whose n^{th} member is a_n.

Imagine a sequence as a succession of projectiles directed towards the bulls-eye of a target, where the center of the bulls-eye is regarded as the limit. We adjust our aim so that the rest of the projectiles will be striking within one inch of that center. The accuracy is the one inch, and that must precede the control. How do we invoke a control? From some projectile on, the rest of them will be withing the requested accuracy. Let's formalize the concept of the limit of a sequence.

Definition: A sequence $\{a_n\}$ has the *limit* L and we write

$$\lim_{n\to\infty} a_n = L \qquad \text{or} \qquad a_n \to L \text{ as } n \to \infty$$

if, for each $\epsilon > 0$ (accuracy) there is a number M (control) so that

$$n > M \;\Rightarrow\; |a_n - L| < \epsilon$$

If we invoke the control by having n larger than M, then our sequence member a_n will be within our predetermined accuracy, $\epsilon > 0$.



In the diagram above, all members of the sequence to the right of the vertical line marked M lie in the shaded region between the two horizontal lines at $L-e$ and $L+e$. That is what is meant by '$|a_n - L| < \epsilon$ for $n > M$'.

When your instructor told you that a sequence is a function whose domain is the natural numbers, or a subset thereof, it is possible that you did not attach as much importance to that idea as you did to the mechanics of dealing with sequences. Maple allows us to define expressions and functions, and it is sometimes confusing as to which we want to use. We will define the sequence $\{a_n\} = \left\{\dfrac{n+2}{3n-1}\right\}$ as an expression and the sequence $\{b_n\} = \left\{\sqrt{n^2+3n} - n\right\}$ as a function to illustrate how they must be handled differently.

Maple Example 1: $\{a_n\} = \left\{\dfrac{n+2}{3n-1}\right\}$

```
> with(student):
> a:=(n+2)/(3*n-1);
```

$$a := \frac{n+2}{3n-1}$$

```
> seq(a,n=1..10);
```

$$\frac{3}{2}, \frac{4}{5}, \frac{5}{8}, \frac{6}{11}, \frac{1}{2}, \frac{8}{17}, \frac{9}{20}, \frac{10}{23}, \frac{11}{26}, \frac{12}{29}$$

```
> limit(a,n=infinity);
```

$$\frac{1}{3}$$

Maple Example 2: $\{b_n\} = \left\{\sqrt{n^2+3n} - n\right\}$

```
> with(student):
> b:=n->sqrt(n^2+3n)-n;
```

$$b := n \to \sqrt{n^2+3n} - n$$

```
> seq(b(n),n=1..10);
```

$$1, \sqrt{10}-2, 3\sqrt{2}-3, 2\sqrt{7}-4, 2\sqrt{10}-5, 3\sqrt{6}-6, \sqrt{70}-7, s\sqrt{22}-8, 6\sqrt{3}-9, \sqrt{130}-10$$

```
> limit(b(n),n=infinity);
```

$$\frac{3}{2}$$

To understand this last limit, consider multiplying b_n by its conjugate, and then dividing by it.

$$\left(\sqrt{n^2+3n} - n\right) \cdot \frac{\sqrt{n^2+3n}+n}{\sqrt{n^2+3n}+n} = \frac{n^2+3n-n^2}{\sqrt{n^2+3n}+n} = \frac{3n}{\sqrt{n^2+3n}+n} \cdot \frac{1/n}{1/n} = \frac{3}{\sqrt{1+3/n}+1} \to \frac{3}{2}$$

with the limit taken as $n \to \infty$.

The sequence $\{a_n\}$ is obtained from an expression whose name is a while the sequence $\{b_n\}$ is obtained by evaluating a function whose name is b. If we had the command `seq(b,n=1..10);` what would we have obtained? The answer - ten b's, because the function b must be evaluated in order for it to have a value. This will be very important in the next section when we will need to consider the term $\dfrac{b_{n+1}}{b_n} = \dfrac{b(n+1)}{b(n)}$. It would be cumbersome and less clear to find $\dfrac{a_{n+1}}{a_n}$ when a is an expression. The command would be `subs(n=n+1,a)/a;`.

C2M10 Problems Using Maple, find the first ten terms of each sequence and the limit of each.

1. $a_n = \left\{\dfrac{n^2}{3^n}\right\}$
2. $b_n = \left\{\dfrac{n^2-3n+4}{5+2n+6n^2}\right\}$
3. $c_n = \left\{\left(1+\dfrac{2}{n}\right)^n\right\}$
4. $d_n = \left\{\left(1-\dfrac{2}{n}\right)^n\right\}$
5. $e_n = \left\{\sqrt{n^2+6n} - n\right\}$

C2M11

Ratio Test

The ratio test is one of the most important tools in the study of infinite series. Its validity is a consequence of what we know about geometric series. For Maple purposes we will define the sequence upon which the series is based as a function of n. So we will use $\displaystyle\sum_{n=1}^{\infty} a_n = \sum_{n=1}^{\infty} a(n)$ which emphasizes that $a_n = a(n)$ is really a function of n.

Example: Discuss the convergence/divergence of the series $\displaystyle\sum_{n=1}^{\infty} \frac{2^{2n+1}}{n\,5^n}$.

We use the ratio test and consider $\dfrac{a_{n+1}}{a_n} = \dfrac{2^{2n+3}}{(n+1)\,5^{n+1}} \cdot \dfrac{n\,5^n}{2^{2n+1}} = \dfrac{2^2\,n}{5\,(n+1)}$. Take the limit $\displaystyle\lim_{n\to\infty} \frac{a_{n+1}}{a_n} = \lim_{n\to\infty} \frac{2^2\,n}{5\,(n+1)} = \frac{4}{5}$ and conclude that the given series converges because the limit is less than one. Now, let's do this same problem using Maple. Note how we define $a_n = a(n)$ **as a function**, but the ratio, r_n, **is an expression**.

Maple Example:

```
> with(student):
> a:=n->2^(2*n+1)/(n*5^n);
```

nth term of series

$$a := n \to \frac{2^{(2n+1)}}{n\,5^n}$$

```
> rn:=a(n+1)/a(n);
```

ratio for series

$$rn := \frac{2^{(2n+3)}n\,5^n}{(n+1)\,5^{(n+1)}2^{(2n+1)}}$$

```
> rn:=simplify(rn);
```

$$rn := \frac{4}{5}\,\frac{n}{n+1}$$

```
> limit(rn,n=infinity);
```

$$\frac{4}{5}$$

C2M11 Problems Use Maple to assist with the ratio test for the given series. Remember to include a concluding remark about the ratio test results.

1. $\displaystyle\sum_{n=1}^{\infty} \frac{(n+1)^2}{3^n\,n!}$ 2. $\displaystyle\sum_{n=1}^{\infty} \frac{(3n)!}{2^{2n}7^n(n!)^3}$ 3. $\displaystyle\sum_{n=1}^{\infty} \frac{n!}{n^n}$

C2M12

Ratio Test for Power Series

We extend our use of Maple to power series by employing the same approach as in the previous section, but realizing that there will be a variable, or parameter, x involved. We may no longer assume that all the terms are positive, so the absolute value **must** be used.

Example: Find the open interval of convergence for the power series $\displaystyle\sum_{n=1}^{\infty} \frac{2^n\,x^n}{n\,3^{n+1}}$.

$\left|\dfrac{a_{n+1}}{a_n}\right| = \left|\dfrac{2^{n+1}x^{n+1}}{(n+1)\,3^{n+2}} \cdot \dfrac{n\,3^{n+1}}{2^n\,x^n}\right| = \dfrac{2}{3}\left|\dfrac{n}{n+1}\right| \cdot |x| \to \dfrac{2}{3}|x|$ with the limit taken as n goes to infinity and x is held constant. When are we guaranteed that this series will converge? When the limit of the ratio test is forced to be less than one. Thus, $\dfrac{2}{3}|x| < 1 \;\Rightarrow\; |x| < \dfrac{3}{2}$. We conclude that the series converges for all x in the open interval $\left(-\frac{3}{2}, \frac{3}{2}\right)$.

Maple Example: Use Maple to find the open interval of convergence of the previous example.

```
> an:=n->(2^n*x^n)/(n*3^(n+1));
```

$$an := n \to \frac{2^n\,x^n}{n\,3^{(n+1)}}$$

```
> rn:=an(n+1)/an(n);
```

$$rn := \frac{2^{(n+1)}\,x^{(n+1)}\,n\,3^{(n+1)}}{(n+1)\,3^{(n+2)}2^n\,x^n}$$

```
> rn:=abs(simplify(rn));
```

$$rn := \frac{2}{3}\left|\frac{xn}{n+1}\right|$$

```
> limit(rn,n=infinity);
```

$$\frac{2}{3}|x|$$

```
> solve(%<1,x);
```

$$\text{RealRange}\left(\text{Open}\left(\frac{-3}{2}\right), \text{Open}\left(\frac{3}{2}\right)\right)$$

So the open interval $(-3/2, 3/2)$ is our answer.

C2M12 Problems: Use Maple to find the open interval of convergence of the given power series.

1. $\displaystyle\sum_{n=1}^{\infty} \frac{n!x^n}{n^n}$ 2. $\displaystyle\sum_{n=1}^{\infty} \frac{nx^n}{(n+1)!}$ 3. $\displaystyle\sum_{n=1}^{\infty} \frac{n!(2n)!x^n}{(3n)!}$ 4. $\displaystyle\sum_{n=1}^{\infty} \frac{(2/3)^n(x+2)^n}{n^2}$

C2M13

Maclaurin and Taylor Series

It is remarkable that knowing about the values of a function and its derivatives at a point provides a means of evaluating the function at points nearby. Maclaurin and Taylor series are that means. Taylor and Maclaurin series are written respectively as

$$f(x) = \sum_{n=0}^{\infty} \frac{f^{(n)}(c)}{n\,!}(x-c)^n \qquad\qquad f(x) = \sum_{n=0}^{\infty} \frac{f^{(n)}(0)}{n\,!}x^n$$

where, by letting $c = 0$, we see that the Maclaurin series is a special case of the Taylor series. The reader is reminded that

$$\begin{aligned} 0! &= 1 \\ 1! &= 1 \cdot 0! = 1 \\ 2! &= 2 \cdot 1! = 2 \\ 3! &= 3 \cdot 2! = 6 \\ 4! &= 4 \cdot 3! = 24 \\ 5! &= 5 \cdot 4! = 120 \\ 6! &= 6 \cdot 5! = 720 \\ 7! &= 7 \cdot 6! = 5040 \\ 8! &= 8 \cdot 7! = 40320 \\ 9! &= 9 \cdot 8! = 362880 \\ 10! &= 10 \cdot 9! = 3628800 \end{aligned}$$

Just for fun, in a Maple worksheet enter `357!;`. The speed with which this computation is done is remarkable.

Frequently it is useful to write out the first few terms of a Taylor series. The result is a *Taylor Polynomial.* For example,

$$T_n = f(a) + f'(a)(x-a) + \frac{f''(a)}{2!}(x-a)^2 + \ldots + \frac{f^{(n)}(a)}{n!}(x-a)^n$$

The Maple syntax for a Taylor polynomial of degree n at $x = a$ is:

```
> taylor(f(x),x=a,n);
```

Every student should grind out a few Taylor series by hand so that they appreciate how the coefficients are determined. But, we can use Maple spreadsheets to accomplish the same thing. A worksheet entitled **Taylor Series Worksheet** follows that does exactly that. We will explain a few subtleties of the 'how's and why's' of this worksheet now.

Begin by defining the function $f(x) = \cos(2x)$ and identifying $a = \pi/6$. Recall that to open the spreadsheet you must click on **Insert** and the **Spreadsheet**. Resize it so that it shows about 10 rows and 4 columns. Enter n in cell **A1**, *function* in cell **B1** and $x = a$ in cell **C1**. Put 0 in **A2**, $f(x)$ in **B2**, and then `eval(~B2,x=a)` in **C2**. Continue by highlighting from **A2** down to **A8**, click on the button on the third row of the menus at the extreme left, insert a step size of 1, and then click **OK**. Move to cell **B3**. Enter `diff(~B2,x)` and then highlight that cell down to **B8**. Click on **Spreadsheet**, **Fill**, and **Down** and the successive derivatives should appear. Highlight from **C2** down to **C8**, and click on **Spreadsheet, Fill, Down** as before. Now the spreadsheet should be complete.

Continue by highlighting cells **C2** down to **C8**, clicking on the **Copy** button, clicking on the command line just below the spreadsheet, and then clicking on the **Paste** button. Important: put a colon at the end of MATRIX material and <Enter>. Convert the matrix to a list as shown, and then note that we have a list of a list of small matrices. So, we must look only at the first entry of our list which is `c[1]`. We want to formulate our coefficients for the Taylor series without the matrix brackets and this is done in our definition of the function b. Check out the numerator of the fraction. We have `c[n+1]` as the n^{th} matrix in c if we start with 0. That matrix has one entry, and to access that number we use `(c[n+1])[1]`.

The coefficients for the Taylor polynomial, T_6, are $b(0), b(1), b(2), b(3), b(4), b(5), b(6)$, which we enter as $b(n)$ in the summation. We used $x-a$ which is easier than $x - Pi/6$. Then, let Maple do the same thing in one line. Compare the coefficients.

Taylor Series Worksheet

```
> restart:    with(student):
```

```
> f:=x->cos(2*x);
```

$$f := x \to \cos(2x)$$

```
> a:=Pi/6:
```

	A	B	C
1	*n*	*function*	$x = \frac{1}{6}\pi$
2	0	$\cos(2x)$	$\frac{1}{2}$
3	1	$-2\sin(2x)$	$-\sqrt{3}$
4	2	$-4\cos(2x)$	-2
5	3	$8\sin(2x)$	$4\sqrt{3}$
6	4	$16\cos(2x)$	8
7	5	$-32\sin(2x)$	$-16\sqrt{3}$
8	6	$-64\cos(2x)$	-32

```
> MATRIX([[1/2], [-sqrt(3)], [-2], [4*sqrt(3)], [8],[-16*sqrt(3)], [-32]]):
```

```
> c:=convert(%,list);
```

$$c := \left[\left[\left[\frac{1}{2}\right], [-\sqrt{3}], [-2], [4\sqrt{3}], [8], [-16\sqrt{3}], [-32]\right]\right]$$

```
> c:=c[1];
```

$$c := \left[\left[\frac{1}{2}\right], [-\sqrt{3}], [-2], [4\sqrt{3}], [8], [-16\sqrt{3}], [-32]\right]$$

```
> b:=n->(c[n+1])[1]/n!;
```

$$b := n- > \frac{c_{n+1_1}}{n!}$$

```
> T6:=sum(b(n)*(x-a)^n,n=0..6);
```

$$T6 := \frac{1}{2} - \sqrt{3}\left(x - \frac{1}{6}\pi\right) - \left(x - \frac{1}{6}\pi\right)^2 + \frac{2}{3}\sqrt{3}\left(x - \frac{1}{6}\pi\right)^3 + \frac{1}{3}\left(x - \frac{1}{6}\pi\right)^4 - \frac{2}{15}\sqrt{3}\left(x - \frac{1}{6}\pi\right)^5 - \frac{2}{45}\left(x - \frac{1}{6}\pi\right)^6$$

```
> taylor(f(x),x=a,7);
```

$$\frac{1}{2} - \sqrt{3}\left(x - \frac{1}{6}\pi\right) - \left(x - \frac{1}{6}\pi\right)^2 + \frac{2}{3}\sqrt{3}\left(x - \frac{1}{6}\pi\right)^3 + \frac{1}{3}\left(x - \frac{1}{6}\pi\right)^4 - \frac{2}{15}\sqrt{3}\left(x - \frac{1}{6}\pi\right)^5 - \frac{2}{45}\left(x - \frac{1}{6}\pi\right)^6 + O\left(\left(x - \frac{1}{6}\pi\right)^7\right)$$

C2M13 Problem: Use Maple and the method illustrated above to find a Taylor polynomial, T_6, for $f(x) = \arctan(x)$ at $a = 1$. Your work should display the spreadsheet, T_6, and the Maple solution.

C2M14

Animation of Taylor Polynomials

From our text we have seen that a function $f(x)$ may be written as the sum of an n^{th} degree polynomial $T_n(x)$ and a remainder term $R_n(x)$ which tends to 0 as n approaches infinity. For an open interval $|x-a| < R$

$$f(x) = T$$



Now we begin the process of setting up our sequence of Taylor Polynomials, which we do by identifying the degrees that we want to see displayed. We will select the first 29 Taylor Polynomials. So, we will start with 1 and increase by 1 until we reach 29. Continuing our worksheet,

```
> nstart:=1:    skip:=1:    frameno:=29:
> framenumbers:=[seq(nstart + skip*i, i=0..frameno-1)]:
```

$$framenumbers := [1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15, 16, 17, 18, 19, 20, 21, 22, 23, 24, 25, 26, 27, 28, 29]$$

Now we will set up our *frameno* (29) Taylor series, convert them to polynomials, and set up the plots of these polynomials. They will appear in red.

```
> A:=display(seq(plot(convert(taylor(f(x),x=0,i),polynom),
   x=a..b,y=c..d,style=line,thickness=2,numpoints=100,
   title="degree = ".i),i=framenumbers),insequence=true):
```

We need matching plots of $f(x)$, which will appear in blue.

```
> B:=animate(f(x),x=a..b,y=c..d,frames=frameno,color=blue):
```

Combine all this into one plot.

```
> display(A,B,view=[a..b,c..d]);
```

degree = 1



You should see the first Taylor Polynomial, which is $y = 1$, displayed with the graph of $f(x)$. Click on the display and a box should appear around the display. Also, a new menu bar appears so that you have what looks like the buttons for a tape player on the bottom line. The button on the left is "stop" and the button next to it is "play". Click on this button.



You should see the rapid display of the Taylor Polynomials with $f(x)$. Now try the next button to the right, and click on it repeatedly.

If you just want to display a few frames of the animation, it is simpler to just identify the numbers of the frames. The lines below could replace the commands above to get a list of frame numbers.

```
> framenumbers:= [1,3,5,7,9,11,13,15,17,19,21,23,25,27,29];
```

$$framenumbers := [1, 3, 5, 7, 9, 11, 13, 15, 17, 19, 21, 23, 25, 27, 29]$$

```
> frameno:=nops(framenumbers);
```

$$frameno := 15$$

The Maple command `nops` returns the number of elements in the list.

C3M14 Problems: 1. Use $-10 \le x \le 10$ and $-10 \le y \le 10$ to set up an animation for the Taylor Polynomials of $\sin(x)$ with $a = 0$.

2. Set up a Taylor Polynomial animation example using $f(x) = \arctan(x)$.

3. Set up a Taylor Polynomial animation example using $f(x) = e^x$. Remember, e^x is `exp(x)` in Maple.

C2M15

Vectors and the Scalar Product

The scalar, "dot", or "inner" product of two vectors is a very important concept that involves the size of each and the amount that each points in the direction of the other. This nebulous statement will have more meaning as you become familiar with vectors. In Maple, use `evalm` when evaluating a vector or matrix expression. Also, you will need the package `linalg`. In earlier versions of Maple one used `dotprod` to find the scalar product of two vectors. However, in Release 5 and 5.1 this command seems to be of better use when working with complex scalars. We are using only the real numbers, $\mathbb{R}$, so it is best to use the Maple command `innerprod` when finding the scalar product.

The concept of finding the projection of one vector on another is very important. First, the projection is a vector. Second, it is a scalar multiple of the vector being projected on, which must be a non-zero vector. When that scalar multiple is subtracted from the first vector the result is a vector that is orthogonal to the vector being projected on. Let's find the projection of $\overrightarrow{v_1} = \langle 2, 1, -3\rangle$ on $\overrightarrow{v_2} = \langle 1, -2, 2\rangle$. We will find the scalar c symbolically and then let Maple do the calculations and verify that it works. We seek c so that $(\overrightarrow{v_1} - c\overrightarrow{v_2}) \cdot \overrightarrow{v_2} = 0$. The *projection of* $\overrightarrow{v_1}$ *on* $\overrightarrow{v_2}$ is the vector $c\overrightarrow{v_2}$. Recall that two vectors are orthogonal (or perpendicular) if their scalar or dot product is 0.

$$
\begin{aligned}
(\overrightarrow{v_1} - c\overrightarrow{v_2}) \cdot \overrightarrow{v_2} &= \overrightarrow{v_1} \cdot \overrightarrow{v_2} - c\overrightarrow{v_2} \cdot \overrightarrow{v_2} = 0 \\
\Longrightarrow \overrightarrow{v_1} \cdot \overrightarrow{v_2} &= c\overrightarrow{v_2} \cdot \overrightarrow{v_2} \\
\Longrightarrow c &= \frac{\overrightarrow{v_1} \cdot \overrightarrow{v_2}}{\overrightarrow{v_2} \cdot \overrightarrow{v_2}}
\end{aligned}
$$

The denominator $\overrightarrow{v_2} \cdot \overrightarrow{v_2}$ cannot be 0 because $\overrightarrow{v_2}$ is assummed to be non-zero. In Maple we have

```
> with(linalg):
> v1:=vector([2,1,-3]):  v2:=vector([1,-2,2]):
> c:=innerprod(v1,v2)/innerprod(v2,v2);
```

$$c := \frac{-2}{3}$$

```
> proj:=evalm(c*v2);
```

$$proj := \left[\frac{-2}{3}, \frac{4}{3}, \frac{-4}{3}\right]$$

```
> ortho:=evalm(v1-proj);
```

$$ortho := \left[\frac{8}{3}, \frac{-1}{3}, \frac{-5}{3}\right]$$

```
> innerprod(ortho,v2);
```

$$0$$

The "norm" or means of measuring the length of a vector that is most important to us is a special case of the so-called p-norm. We demonstrate for the vector $\overrightarrow{v} = \langle 2, -3\rangle$.

$$||\overrightarrow{v}||_p = (|2|^p + |-3|^p)^{1/p}$$

This has meaning if $p = 2$ and we will always use that value. So we will suppress the subscripts. This explanation is provided so that the '2' in the Maple command makes some sense. If you divide a non-zero vector by its length you have a vector of length one, called a "unit" vector. This process is called "normalizing" a vector. The vector $\overrightarrow{u}$ below is a unit vector in the direction of $\overrightarrow{v}$.

$$||\overrightarrow{v}||_2 = ||\overrightarrow{v}|| = ||\langle 2, -3\rangle|| = \sqrt{(2)^2 + (-3)^2} = \sqrt{13} \qquad \overrightarrow{u} = \frac{1}{\sqrt{13}}\langle 2, -3\rangle$$

Returning to the Maple worksheet that we started above, let's compute $||\overrightarrow{v_2}||$ and normalize $\overrightarrow{v_2}$. The Maple command `normalize` assumes that the '2' norm is being used.

```
> lengthv2:=norm(v2,2);
```

$$lengthv2 := 3$$

```
> u2:=normalize(v2);
```

$$u2 := \left[\frac{1}{3}, \frac{-2}{3}, \frac{2}{3}\right]$$

Recall that $\overrightarrow{a} \cdot \overrightarrow{b} = ||\overrightarrow{a}||\,||\overrightarrow{b}|| \cos\theta$, where θ is the angle between $\overrightarrow{a}$ and $\overrightarrow{b}$. Obviously, for non-zero vectors $\overrightarrow{a}$ and $\overrightarrow{b}$ we can solve for $\cos\theta$ and thereby determine θ.

$$\cos\theta = \frac{\overrightarrow{a} \cdot \overrightarrow{b}}{||\overrightarrow{a}||\,||\overrightarrow{b}||}$$

Maple saves us the work with the command "angle".

```
> theta:=angle(v1,v2);
```

$$\theta := \pi - \arccos\left(\frac{1}{21}\sqrt{14}\sqrt{9}\right)$$

```
> theta:=evalf(theta);
```

$$\theta := 2.134738969$$

```
> theta1:=evalf(theta*180/Pi);
```

$$\theta 1 := 122.3115332$$

This makes sense in the following diagram where we see that θ must be an obtuse angle since the projection is opposite to the direction of $\vec{v2}$.



The projection of $\vec{v1}$ on $\vec{v2}$

A line between two points P and Q is parameterized by $\alpha(t) = (1-t)\vec{P} + t\vec{Q}$ for $0 \leq t \leq 1$. We will use this in combination with `spacecurve` to plot lines.

Example: Plot the coordinate axes and the line segment from $P(4, -1, 2)$ to $Q(1, 4, 4)$.

This task may be accomplished in two distinct ways. The way which is displayed below is direct and uses the basics. Instead of defining the function `alpha`, and plotting it using `spacecurve`, one could use a command in `plottools` named `line`. The syntax would look like

```
> with(plottools):
> LPQ:=line([4,-1,2],[1,4,4],color=red):
```

It suits our purposes here to have you learn how to set up a function like `alpha` because in Calculus III this is a skill which will be essential.

```
> with(plots):
> P:=vector([4,-1,2]):  Q:=vector([1,4,4]):
> xaxis:=spacecurve([t,0,0],t=0..3,color=black):
> yaxis:=spacecurve([0,t,0],t=0..3,color=black):
> zaxis:=spacecurve([0,0,t],t=0..3,color=black):
> alpha:=evalm((1-t)*P+t*Q);
```

$$\alpha := [4 - 3t, -1 + 5t, 2 + 2t]$$

```
> LPQ:=spacecurve(alpha,t=0..1,color=red):
> LP:=spacecurve(evalm(t*P),t=0..1,color=blue):
> LQ:=spacecurve(evalm(t*Q),t=0..1,color=green):
> display(xaxis,yaxis,zaxis,LPQ,LP,LQ);
```

$$-1$$

```
> innerprod(N,R);
```

$$-1$$

We used $\overrightarrow{P}$ as the known vector in the plane. After we found the equation we took the scalar product of $\overrightarrow{N}$ with $\overrightarrow{Q}$ and $\overrightarrow{R}$ just to check and see if we got the same value. With Maple doing the calculations for us no error occurred, but when doing this very basic problem with pencil and paper it is important to check your answer.

Now let's find the area of $\triangle PQR$. The two vectors $\overrightarrow{QP}$ and $\overrightarrow{QR}$ generate a parallelogram whose area is $\|\overrightarrow{QP} \times \overrightarrow{QR}\| = \|\overrightarrow{N}\|$ and the triangle is one-half of that parallelogram.

```
> area:=norm(N,2)/2;;
```

$$area := \frac{1}{2}\sqrt{41}$$

A line in $\mathbb{R}^3$ requires a direction and a point on the line. If $\overrightarrow{N} = \langle a, b, c\rangle$ is the direction vector, $\overrightarrow{X_0} = \langle x_0, y_0, z_0\rangle$ is a specific point on the line, t is a scalar (parameter) and $\overrightarrow{X} = \langle x, y, z\rangle$ is any point on the line, then the parametric equation of the line is

$$\overrightarrow{X} = \overrightarrow{X_0} + t\,\overrightarrow{N} \qquad \text{or equivalently} \qquad \begin{aligned} x &= x_0 + at \\ y &= y_0 + bt \\ z &= z_0 + ct \end{aligned}$$

If we want a parametric equation of a line **normal** (perpendicular, orthogonal) to our plane above and passing through P then

```
> nline:=evalm(X=P+t*N);
```

$$nline := [x, y, z] = [2 + 2t, 1 + t, -1 + 6t]$$

To obtain a parametric equation for the line through P in the direction of Q, we simply use $\overrightarrow{QP}$ as the direction vector.

```
> line:=evalm(X=P+t*QP);
```

$$line := [x, y, z] = [2 + 3t, 1, -1 - t]$$

Let's turn our attention to the distance from a point to a plane. Suppose $ax + by + cz = d$ is an equation for the plane and P is a point. You know that the coefficients a, b, c are attitude numbers for the plane and that $\overrightarrow{N} = \langle a, b, c\rangle$ is a normal vector. Find any point X_0 which lies in the plane (X_0 satisfies the equation). Form the vector $\overrightarrow{\alpha} = \overrightarrow{P - X_0}$. The distance from the point P to the plane will be the length of the projection of $\overrightarrow{\alpha}$ on $\overrightarrow{N}$.

$$\text{distance} = \|c\overrightarrow{N}\| = \left\|\frac{\overrightarrow{\alpha}\cdot\overrightarrow{N}}{\overrightarrow{N}\cdot\overrightarrow{N}}\overrightarrow{N}\right\| = \frac{|\overrightarrow{\alpha}\cdot\overrightarrow{N}|}{\|\overrightarrow{N}\|^2}\|\overrightarrow{N}\| = \frac{|\overrightarrow{\alpha}\cdot\overrightarrow{N}|}{\|\overrightarrow{N}\|}$$

Please note the diagram for this later in the section.

Example 2: Find the distance from $P = (4, -3, 5)$ to the plane with equation $2x + y + 6z = -1$.

We observe that $X_0 = (-1, 1, 0)$ lies in the plane and that $\overrightarrow{N} = \langle 2, 1, 6\rangle$ is a normal vector to this plane.

```
> N:=vector([2,1,6]):  P:=vector([4,-3,5]):  X0:=vector([-1,1,0]):
> alpha:=evalm(P-X0);
```

$$\alpha := [5, -4, 5]$$

```
> distance:=abs(innerprod(alpha,N))/norm(N,2);
```

$$distance := \frac{36}{41}\sqrt{41}$$

Let's turn our attention to finding the distance from a point to a line. If P is the point and $\overrightarrow{X} = \overrightarrow{X_0} + t\,\overrightarrow{N}$ is a parametric equation for the line, then it is easy to find X_0 which is a specific point on the line. Let $\overrightarrow{\alpha} = P - X_0$. An examination of the diagram and a little trigonometry shows that the distance $d = \|\overrightarrow{\alpha}\|\sin\theta$ where θ is the angle between $\overrightarrow{\alpha}$ and $\overrightarrow{N}$. Recall that for vectors $\overrightarrow{A}$ and $\overrightarrow{B}$, $\|\overrightarrow{A} \times \overrightarrow{B}\| = \|\overrightarrow{A}\|\,\|\overrightarrow{B}\|\sin\theta$. This leads to

$$d = \|\overrightarrow{\alpha}\|\sin\theta = \|\overrightarrow{\alpha}\|\sin\theta\frac{\|\overrightarrow{N}\|}{\|\overrightarrow{N}\|} = \frac{\|\overrightarrow{\alpha}\|\,\|\overrightarrow{N}\|\sin\theta}{\|\overrightarrow{N}\|} = \frac{\|\overrightarrow{\alpha}\times\overrightarrow{N}\|}{\|\overrightarrow{N}\|}$$

Please note the similarities and contrasts between our two distance problems. Diagrams for each follow.



distance from point P to plane



distance from point P to line

Example 3: Find the distance from the point $P = (4, -3, 5)$ to the line with parametric equations

$$\begin{aligned} x &= -1 - 3t \\ y &= 1 + 5t, \\ z &= \quad + 4t \end{aligned}$$

We identify $\overrightarrow{N} = \langle -3, 5, 4 \rangle$ and $X_0 = (-1, 1, 0)$. In Maple

```
> N:=vector([-3,5,4]):  X0:=vector([-1,1,0]):  P:=vector([4,-3,5]):
> alpha:=evalm(P-X0);
```

$$\alpha := [5, -4, 5]$$

```
> CP:=crossprod(alpha,N);
```

$$CP := [-41, -35, 13]$$

```
> d:=norm(CP,2)/norm(N,2);
```

$$d := \frac{1}{2}\sqrt{123}\sqrt{2}$$

Exercises (pencil and paper):

1. Find an equation for the plane that contains the points $P(2, 3, -1)$, $Q(-1, 1, 4)$, $R(0, 1, 3)$ and the area of the triangle that they form. Find parametric equations for the line through P and Q.

2. Find an equation for the plane that contains the points $P(1, -1, 1)$, $Q(0, 2, -1)$, $R(3, 3, -1)$ and the area of the triangle that they form. Find parametric equations for the line through P and Q.

3. Find the distance from the point $P(3, 3, -2)$ to the plane whose equation is $2x - 3y + 2z = 7$.

4. Find the distance from the point $Q(3, 2, -1)$ to the plane whose equation is $x + 4y - z = 6$.

5. Find the distance from the point $P(3, 3, -2)$ to the line through $Q(1, 2, -1)$ and $R(3, -1, 5)$ and find parametric equations for the line.

6. Find the distance from the point $P(-2, 1, 3)$ to the line through $Q(0, 2, -1)$ and $R(-1, 4, 1)$ and find parametric equations for the line.

C2M16 Problems Use Maple to solve the problems 1, 3, and 5 above.

SUMMARY OF MAPLE COMMANDS

Using the package ‘student’

Maple Command	**Output**
`area:=Pi*r^2;`	$area := \pi r^2$ Sets πr^2 as an expression in r named *area*
`f:=x->sqrt(4+9*x^2);`	$f := x \to \sqrt{4+9x^2}$ Sets $\sqrt{4+9x^2}$ as a function named $f(x)$
`subs(r=5,area);`	25π Substitutes 5 into the expression *area* as r
`f(1);`	$\sqrt{13}$ Evaluates the function f at 1
`eval(expr,x=a);`	Evaluates the expression *expr* at $x = a$
`plot(area,r=1..4);`	Produces a plot of expression *area*, r ranges from 1 to 4
`plot(f(x),x=-2..5);`	Produces a plot of function $f(x)$, x ranges from -2 to 5
`diff(area,r);`	$2\pi r$ The derivative of the expression *area* with respect to r
`diff(f(x),x);`	$9\frac{x}{\sqrt{4+9x^2}}$ The derivative of the function $f(x)$ with respect to x
`int(area,r);`	$\frac{1}{3}\pi r^3$ The anti-derivative of the expression *area* with respect to r
`int(area,r=1..4);`	21π The definite integral of *area* from 1 to 4.
`evalf(%);`	65.97344573 The decimal form of the previous expression, 21π
`Int(area,r);`	$\int \pi\, r^2\, dr$ The inert expression whose value is the antiderivative of *area*
`sum(f(k),k=n1..n2);`	Evaluates the sum $\sum_{k=n1}^{k=n2} f(k)$ of $f(k)$ from $n1$ to $n2$
`leftbox(f(x),x=a..b,n);`	Displays the Riemann sum graphically for $f(x)$ over the interval $[a,b]$ using n intervals and left endpoints
`rightsum(f(x),x=a..b,n);`	Displays the Riemann sum in summation form for $f(x)$ over the interval $[a,b]$ using n intervals and right endpoints

Using the package ‘plots’

`spacecurve([f(t),g(t),h(t)],t=a..b);`	Displays the three-dimensional curve defined parametrically by three expressions in t, which ranges from a to b
`plot3d(expr,var1=a..b,var2=c..d);`	Displays a surface for an expression in two variables
`plot3d([expr1,expr2,expr3],var1=a..b, var2=c..d);`	Displays a surface parametrically as a function of two variables
`cylinderplot(expr,var1=a..b,var2=c..d);`	Displays $z =$expr in terms of $r =$var1 and $\theta =$var2
`polarplot(expr,var=a..b);`	Displays a polar coordinate plot of $r =$expr in terms of $\theta =$var
`sphereplot(expr,var1=a..b,var2=c..d);`	Displays $\rho =$expr in terms of $\theta =$var1 and $\phi =$var2
`cylinderplot([expr1,expr2,expr3], var1=a..b,var2=c..d);`	Displays a 3-d cylindrical plot parametrically, (r, θ, z)
`polarplot([expr1,expr2,var=a..b]);`	Displays a parametric polar coordinate plot of $r =$expr1 and $\theta =$expr2
`sphereplot([expr1,expr2,expr3], var1=a..b,var2=c..d);`	Displays a 3-d spherical plot parametrically, (ρ, θ, ϕ)
`implicitplot(equation,var1=a..b, var2=c..d);`	Displays a 2-d implicit plot of equation
`implicitplot3d(eqn,var1=a..b, var2=c..d,var3=e..f);`	Displays a 3-d plot for the equation eqn
`display({Plot1,Plot2,..,Plotn});`	Displays n plots on the same coordinate system

Using the package ‘linalg’

`A:=vector([a,b,c]);`	Assigns the name A to the vector $\langle a, b, c\rangle$
`A[1]; A[2]; A[3];`	The components of the vector A are displayed
`F:=vector([expr1,expr2,expr3]);`	Assigns the name F to the vector expression
`map(diff,F,t);`	Gives the derivative of the vector $\overrightarrow{F}$ with respect to t
`innerprod(A,B);`	Finds $\overrightarrow{A} \cdot \overrightarrow{B}$, the scalar product of vectors $\overrightarrow{A}$ and $\overrightarrow{B}$
`crossprod(A,B);`	Finds $\overrightarrow{A} \times \overrightarrow{B}$, the cross product of vectors $\overrightarrow{A}$ and $\overrightarrow{B}$
`evalm(A+c*B);`	Finds $A + cB$ for vectors $\overrightarrow{A}$ and $\overrightarrow{B}$, and scalar c
`hessian(f,[x,y]);`	Finds the hessian matrix $\begin{bmatrix} f_{xx} & f_{xy} \\ f_{yx} & f_{yy} \end{bmatrix}$ for the function f of x and y
`det(A);`	Finds the determinant of the square matrix A
`subs({x=a,y=b},op(H));`	Substitutes values for x and y into components of matrix/vector H